

Tabela de conteúdos

1. [Introduction](#) 1.1
2. [Visão geral](#) 1.2
 1. [Instalação](#) 1.2.1
 2. [Opções](#) 1.2.2
 3. [Empacotamento](#) 1.2.3
 4. [Ferramentas](#) 1.2.4
3. [Sintaxe](#) 1.3
 1. [Formatação](#) 1.3.1
 2. [Importações](#) 1.3.2
 3. [Entradas](#) 1.3.3
 4. [Tipos](#) 1.3.4
 1. [Any](#) 1.3.4.1
 2. [Void](#) 1.3.4.2
 3. [Boolean](#) 1.3.4.3
 4. [Number](#) 1.3.4.4
 5. [Text](#) 1.3.4.5
 6. [Method](#) 1.3.4.6
 7. [List](#) 1.3.4.7
 8. [Scope](#) 1.3.4.8
 9. [Error](#) 1.3.4.9
 5. [Controle de fluxo](#) 1.3.5
 6. [Loops](#) 1.3.6
4. [Bibliotecas](#) 1.4
 1. [async](#) 1.4.1
 2. [color](#) 1.4.2
 3. [console](#) 1.4.3
 4. [curses](#) 1.4.4
 5. [failure](#) 1.4.5
 6. [file](#) 1.4.6
 7. [http](#) 1.4.7
 8. [math](#) 1.4.8
 9. [sdk](#) 1.4.9
 10. [system](#) 1.4.10
 11. [time](#) 1.4.11
 12. [zCode](#) 1.4.12
 13. [type_](#) 1.4.13
 1. [Void](#) 1.4.13.1
 2. [Boolean](#) 1.4.13.2
 3. [Number](#) 1.4.13.3
 4. [Text](#) 1.4.13.4
 5. [Method](#) 1.4.13.5
 6. [List](#) 1.4.13.6
 7. [Scope](#) 1.4.13.7
 8. [Error](#) 1.4.13.8
 14. [extra_](#) 1.4.14
 1. [csv](#) 1.4.14.1
 2. [Date](#) 1.4.14.2
 3. [Duration](#) 1.4.14.3
 4. [elapsed](#) 1.4.14.4
 5. [HashMap](#) 1.4.14.5
 6. [hex](#) 1.4.14.6
 7. [json](#) 1.4.14.7
 8. [Logger](#) 1.4.14.8
 9. [mathex](#) 1.4.14.9
 10. [Memo](#) 1.4.14.10
 11. [regex](#) 1.4.14.11
 12. [Sound](#) 1.4.14.12
 13. [util](#) 1.4.14.13
 14. [xml](#) 1.4.14.14
 15. [Comandos embutidos](#) 1.4.15

Introduction

 FatScript logo

Olá Mundo

```
_ <- fat.console  
log('Olá Mundo')
```

Início Rápido

Vá diretamente para a documentação:

- [Visão geral](#)
- [Sintaxe da linguagem](#)
- [Bibliotecas padrão](#)

Executando seu código

Você pode executar o FatScript usando o interpretador `fry` ou o playground na web.

Interpretador Fry

Para execução local, utilize o interpretador `fry`. Para detalhes sobre sua instalação e use, consulte a seção de [configuração](#).

Playground na Web (beta)

Para testes rápidos e convenientes, execute seu código diretamente no [FatScript Playground](#). O playground oferece um REPL e uma interface intuitiva que permite carregar scripts a partir de um arquivo.

Tutoriais

Mergulhe em nossos tutoriais imersivos, insights por trás dos bastidores e tópicos relacionados no [canal do YouTube FatScript](#).

Comunidade

Junte-se à [comunidade do FatScript no Reddit](#) para se conectar com outros programadores e discutir tudo sobre o FatScript!

Doações

Você achou o FatScript útil e gostaria de agradecer?

[Compre-me um café](#)

Licença

[GPLv3](#) © 2022-2023 Antonio Prates

[fatscript.org](#)

Visão geral

Visão Geral

FatScript é uma linguagem de programação leve e interpretada projetada para criar aplicativos baseados em console. Ela enfatiza a simplicidade, facilidade de uso e conceitos de programação funcional.

Livre e de código aberto

`fatscript/fry` é um projeto de código aberto que incentiva a colaboração e o compartilhamento de conhecimento. Nós convidamos os desenvolvedores a [contribuir](#) para o projeto e nos ajudar a melhorá-lo com o tempo.

Conceitos chave

- Gerenciamento automático de memória por coleta de lixo (GC)
- Combinações simbólicas de caracteres para uma sintaxe minimalista
- REPL (Read-Eval-Print Loop) para testes rápidos de expressões
- Suporte para sistema de tipos, herança e subtipagem por meio de aliases
- Suporte para programação imutável e métodos passáveis (como valores)
- Manter se simples e intuitivo, sempre que possível

Conteúdo desta seção

- [Instalação](#): como instalar o interpretador de FatScript
- [Opções](#): como personalizar a execução
- [Empacotamento](#): como empacotar um aplicativo FatScript
- [Ferramentas](#): visão geral de algumas ferramentas e recursos extras

Limitações e desafios

Embora o FatScript seja projetado para ser simples e intuitivo, ele ainda é uma linguagem relativamente nova e pode não ser adequado para todos os casos de uso. Por exemplo, pode ter desempenho inferior em comparação com linguagens de programação mais maduras ao lidar com cargas de trabalho complexas ou tarefas de computação de alto desempenho.

Instalação

Instalação

Para começar a "fritar" seu código "gordo", você precisará de um interpretador para a linguagem de programação FatScript.

fry, o Interpretador Fatscript

[fry](#) é um interpretador e ambiente de execução gratuito para FatScript. Você pode instalá-lo em sua máquina seguindo as instruções a seguir.

Instruções

fry é projetado para GNU/Linux, mas também pode funcionar em [outros sistemas operacionais](#).

Para distribuições baseadas em Arch, instale através do pacote AUR [fatscript-fry](#).

Para outras distribuições, experimente o script de instalação automática:

```
curl -sSL https://gitlab.com/fatscript/fry/raw/main/get_fry.sh -o get_fry.sh;
bash get_fry.sh || sudo bash get_fry.sh
```

Ou, para instalar `fry` manualmente:

- Clone o repositório:

```
git clone --recursive https://gitlab.com/fatscript/fry.git
```

- Depois, execute o script de instalação:

```
cd fry
./install.sh
```

a instalação manual pode copiar o binário `fry` para a pasta `$HOME/.local/bin`, alternativamente, use `sudo` para instalá-lo em `/usr/local/bin/`

- Verifique se o `fry` foi instalado, executando:

```
fry --version
```

Dependências

Se a instalação falhar, podem estar faltando algumas dependências. `fry` requer `git`, `gcc` e `libcurl` para compilar. Por exemplo, para instalar essas dependências no Debian/Ubuntu, execute:

```
apt update
apt install git gcc libcurl4-openssl-dev
```

Suporte de Sistema Operacional

`fry` é primordialmente projetado para GNU/Linux, mas também é acessível em outros sistemas operacionais:

Android

Se você estiver no Android, pode instalar o `fry` via [Termux](#). Basta instalar as dependências necessárias da seguinte maneira:

```
pkg install git clang
```

Em seguida, você pode seguir as instruções padrão de instalação do `fry`.

ChromeOS

Se você estiver usando o ChromeOS, pode habilitar o suporte ao Linux seguindo as instruções [aqui](#).

MacOS

Se você estiver usando o MacOS, precisará ter as [Command Line Tools](#) instaladas.

iOS

Se você estiver usando o iOS, poderá usar o fry via [iSH](#). Primeiro, instale as dependências necessárias:

```
apk add bash gcc libc-dev curl-dev
```

Em seguida, de acordo com esta [discussão](#), configure o git para funcionar corretamente, assim:

```
wget https://dl-cdn.alpinelinux.org/alpine/v3.11/main/x86/git-2.24.4-r0.apk
apk add ./git-2.24.4-r0.apk
git config --global pack.threads "1"
```

Windows

Se você estiver usando o Windows, poderá usar o fry via [Windows Subsystem for Linux \(WSL\)](#).

Imagem Docker

fry também está disponível como uma [imagem docker](#):

```
docker run --rm -it fatscript/fry
```

Para executar um arquivo FatScript com o docker, use o seguinte comando:

```
docker run --rm -it -v ~/project:/app fatscript/fry prog.fat
```

substitua ~/project pelo caminho para o seu arquivo FatScript

Solução de problemas

Se você encontrar qualquer problema ou bug ao usar o fry, por favor [abra uma "issue"](#).

Opções

Opções

Com esta descrição dos modos e parâmetros disponíveis, você descobrirá que o `fry` tem várias especiarias guardadas na manga para você temperar a execução do seu código.

Argumentos de linha de comando

A interface CLI oferece alguns modos de operação:

- `fry [OPÇÕES] read-eval-print-loop (REPL)`
- `fry [OPÇÕES] ARQUIVO [ARGS]` executar um arquivo FatScript
- `fry [OPÇÕES] -b/-o ENTRADA SAÍDA` criar um bundle
- `fry [OPÇÕES] -f ARQUIVO...` formatar arquivos de código-fonte do FatScript

Aqui estão os parâmetros de opção disponíveis:

- `-a`, `--ast` exibir apenas a árvore sintática abstrata
- `-b`, `--bundle` salvar bundle em arquivo de saída (implica em `-p`)
- `-c`, `--clock` habilitar o registro de tempo e estatísticas (benchmark)
- `-d`, `--debug` habilitar logs de debug (implica em `-c`)
- `-e`, `--error` continuar em caso de erro (alternar)
- `-f`, `--format` formatar arquivos de código-fonte do FatScript
- `-h`, `--help` exibir ajuda e sair
- `-i`, `--interactive` iniciar REPL após execução do arquivo
- `-k`, `--stack #` define a profundidade da pilha (contagem de frames)
- `-m`, `--meta` exibir informações sobre a build
- `-n`, `--nodes #` definir limite de memória (contagem de nós)
- `-o`, `--obfuscate` ofuscar o bundle (implica em `-b`)
- `-p`, `--probe` realizar análise estática (teste seco)
- `-s`, `--save` armazenar sessão do REPL em `repl.fat`
- `-v`, `--version` exibir número da versão e sair
- `-w`, `--warranty` exibir isenção de responsabilidade e sair

Observe que, quando no modo REPL ou ao usar `--probe`, a opção `-e` (continuar em caso de erro) é ativada por padrão.

Gerenciamento de memória

`fry` gerencia a memória automaticamente sem pré-reserva. Você pode limitar o uso da memória especificando o número de nós com as opções da CLI:

- `-n <count>` para uma contagem exata de nós
- `-n <count>k` para kilonós, contagem * 1000
- `-n <count>m` para meganós, contagem * 1000000

Por exemplo, `fry -n 5k meuPrograma.fat` restringe o aplicativo a 5000 nós.

O coletor de lixo (GC) é executado automaticamente quando restam 256 nós antes que o limite final de memória seja atingido (premonição do GC). Você também pode invocar o GC a qualquer momento chamando o método `runGC` da [biblioteca system](#) desde a thread principal.

Usar uma contagem de nós negativa desativa o coletor de lixo, mas ainda define o limite de memória para o valor absoluto, o que pode ser útil para fins de depuração.

Estimativa de bytes (x64)

Cada nó em uma plataforma de 64 bits usa aproximadamente ~200 bytes. O tamanho real do nó depende dos dados que ele contém. Por exemplo, o limite padrão é 10 milhões de nós, seu programa pode chegar a usar cerca de 2 GB de RAM ao atingir o limite padrão.

Use a opção `-c` ou `--clock` para imprimir as estatísticas de execução e ter uma melhor compreensão de como seu programa está se comportando na prática.

Verificação de tempo de execução

Existem dois [comandos embutidos](#) para verificar o uso de memória em tempo de execução:

- `$nodesUsage` nós alocados no momento ($O(1)$)
- `$bytesUsage` bytes alocados no momento ($O(n)$)

verificar os bytes alocados no momento é uma operação cara, pois precisa percorrer todos os nós para verificar o tamanho real de cada um

Tamanho da pilha

A profundidade máxima da pilha é definida em `parameters.h`, no entanto, você pode personalizar o tamanho da pilha até certo ponto usando opções da linha de comando (CLI):

- `-k <count>` para um número exato de frames
- `-k <count>k` para kibiframes, `count * 1024`

Arquivo "run commands"

Na inicialização, `fry` procura um arquivo `.fryrc` no mesmo caminho do arquivo do programa e, se não encontrado, também no diretório de atual. Se encontrado, é executado como uma fase de "pré-cozimento" para configurar o ambiente para a execução do programa.

Gerenciamento de memória com `.fryrc`

Você pode usar o arquivo `.fryrc` para definir o limite de memória para seu projeto sem precisar especificá-lo como argumento da CLI. Para fazer isso, você pode usar o método `setMem` fornecido pela [biblioteca system](#), assim:

```
_ <- fat.system
setMem(64000) # define 64k nós como limite de memória
```

Detalhes de inicialização

As opções da linha de comando são aplicadas primeiro, exceto pelo limite de memória. Durante a fase de pré-cozimento, o `fry` usa o limite padrão de 10 milhões de nós, independentemente da opção da linha de comando. Se você definir um limite de memória no arquivo `.fryrc`, esse limite terá efeito a partir desse ponto e substituirá a opção da linha de comando para toda a execução. Se o arquivo `.fryrc` não definir um limite de memória, a opção da linha de comando terá efeito após a fase de pré-cozimento.

O escopo de pré-cozimento é invisível por padrão. Após a execução do arquivo `.fryrc`, um escopo zerado é fornecido para o seu programa, o que permite testar seu código com um limite muito baixo de nós ao usar um arquivo `.fryrc` sem afetar a contagem de nós. Isso também impede que o namespace `.fryrc` entre em conflito com o escopo global do seu programa. No entanto, se você quiser manter as entradas declaradas no `.fryrc` no escopo global para fins de configuração, pode chamar o comando embutido `$keepDotFry` em algum lugar do arquivo `.fryrc`.

Outro uso possível, além de configurar o limite de memória, é pré-carregar as importações comuns, por exemplo, os tipos padrão:

```
$keepDotFry
_ <- fat.type._
```

Veja também

- [Comandos embutidos](#)
- [Biblioteca system](#)

Empacotamento

Empacotamento

O Fry oferece uma ferramenta integrada de empacotamento para código FatScript.

Utilização

Para agrupar seu projeto em um único arquivo a partir do ponto de entrada, execute:

```
fry -b sweet mySweetProject.fat
```

Este processo consolida todas as importações (exceto [caminhos literais](#)) e remove espaços desnecessários, melhorando os tempos de carregamento:

- Adiciona um [shebang](#) ao código empacotado
- Recebe o atributo de execução para o modo de arquivo

A seguir, você pode executar seu programa:

```
./sweet
```

Ofuscação

Para uma ofuscação opcional, use `-o`:

```
fry -o sweet mySweetProject.fat # cria o pacote ofuscado
./sweet                        # executa seu programa da mesma maneira
```

Ao distribuir por meio de hosts públicos, considere [definir uma chave personalizada](#) com um `.fryrc` local. Apenas o cliente deve ter acesso a esta chave para proteger o fonte.

A ofuscação usa o [zCode](#) para codificação, garantindo uma decodificação rápida. Para um tempo de carregamento ótimo, prefira `-b` se a ofuscação não for essencial.

Considerações

Embora essas considerações geralmente sejam inconsequentes para projetos pequenos, o empacotamento de projetos maiores pode exigir uma organização adicional.

A Ordem Importa

- As importações são deduplicadas e incluídas com base na ordem de sua primeira aparição.
- Como resultado, a sequência em que você importa seus arquivos desempenha um papel crítico no resultado final agrupado. Se dois ou mais arquivos importarem o mesmo módulo, apenas a primeira importação encontrada será incluída no pacote.

Escopo das Importações Nomeadas

- O código não agrupado pode promover importações nomeadas para o escopo global, um comportamento que não é replicado no código agrupado.
- Certifique-se de que as importações nomeadas permaneçam acessíveis nos escopos necessários. Sempre valide o seu código empacotado.

Ferramentas

Ferramentas

Aqui estão algumas dicas que podem melhorar sua experiência de programação com FatScript.

Formatação do código-fonte

Suporte nativo

Você pode aplicar a indentação automática ao seu código fonte usando o seguinte comando:

```
fry -f mySweetProgram.fat
```

Extensão do Visual Studio Code

Para adicionar suporte de formatação de código ao VS Code, você pode instalar a extensão [fatscript-formatter](#). Abra o Quick Open do VS Code (Ctrl+P), cole o seguinte comando e pressione enter:

```
ext install aprates.fatscript-formatter
```

o fry precisa estar instalado em seu sistema para que essa extensão funcione

Realce de sintaxe

Extensão do Visual Studio Code

Para adicionar destaque de sintaxe do FatScript ao VS Code, você pode instalar a extensão [fatscript-syntax](#). Abra o Quick Open do VS Code (Ctrl+P), cole o seguinte comando e pressione enter:

```
ext install aprates.fatscript-syntax
```

Você também pode encontrar e instalar essas extensões no Marketplace de Extensões do VS Code.

Arquivo de sintaxe do Nano

Para instalar o realce de sintaxe do FatScript no nano, siga estes passos:

1. Baixe o arquivo `fat.nanorc` [daqui](#).
2. Copie o arquivo `fat.nanorc` para o diretório de sistema do nano:

```
sudo cp fat.nanorc /usr/share/nano/
```

Se o realce de sintaxe não for habilitado automaticamente, talvez você precise habilitá-lo explicitamente em seu arquivo `.nanorc`. Consulte as instruções na [Wiki do Arch Linux](#) para mais informações.

Após a instalação do destaque de sintaxe, você também pode usar o formatador de código no nano com a seguinte sequência de atalhos:

- Ctrl+T Executar; e em seguida...
- Ctrl+O Formatador

Outras dicas

Navegação de arquivos no console

Para navegar pelas pastas do seu projeto a partir do terminal, você pode experimentar usar um gerenciador de arquivos do console como o [ranger](#), combinado com o nano. Defina-o como o editor padrão para o ranger adicionando a seguinte linha ao seu arquivo `~/ .bashrc`:

```
export EDITOR="nano"
```

Sintaxe

Sintaxe

Nas seguintes páginas, você encontrará informações sobre os aspectos centrais da escrita de código FatScript, utilizando tanto os recursos básicos da linguagem quanto os recursos avançados do sistema de tipos e bibliotecas padrão.

Tópicos abordados

- [Formatação](#): como formatar corretamente o código FatScript
- [Imports](#): como importar bibliotecas para o seu código
- [Entries](#): entendendo o conceito de entradas e escopos
- [Tipos](#): um guia para o sistema de tipos FatScript
- [Controle de fluxo](#): controlando a execução do programa com condicionais
- [Loops](#): utilizando intervalos, map-over e while loops

Formatação

Formatação

No FatScript, espaços em branco e indentação são irrelevantes, porém são muito bem-vindos para tornar o código mais legível e fácil de entender.

Espaços em branco

- Um caractere de nova linha (`\n`) indica o final de uma expressão, exceto quando:
 - o último token na linha é um operador
 - o primeiro token da próxima linha é um operador não-unário
 - usando parênteses para agrupar expressões
- Expressões podem estar na mesma linha se separadas por vírgula (,) ou ponto-e-vírgula (;)

Comentários

Comentários começam com `#` e são terminados por uma nova linha:

```
a = 5 # este é um comentário
```

Nota

FatScript não suporta comentários multi-linhas no momento. Além disso, literais de texto podem acabar como um valor de retorno válido se deixados como a última linha restante, devido à funcionalidade de [auto-retorno](#). Portanto, é recomendável se ater ao formato de comentário de linha única.

Veja também

- [Autoformatador de código fonte](#)

Importações

Importações

Vamos desvendar a arte de importar arquivos e bibliotecas no FatScript! Por quê? Bem, porque nesta linguagem você pode importar sempre que seu coração desejar, simplesmente usando uma seta para a esquerda <-.

Sintaxe de ponto

Para usar importações com sintaxe de ponto, os nomes dos arquivos e pastas do projeto não devem começar com dígito nem conter símbolos.

você pode forçar qualquer caminho que desejar usando [caminhos literais](#)

Importação nomeada

Para importar arquivos, use a extensão `.fat` para nomes de arquivo (ou nenhuma extensão), mas omita a extensão na declaração de importação. Aqui está um exemplo:

```
ref <- nomeDoArquivo
```

se ambos os arquivos `x` e `x.fat` existirem, o último terá precedência

Para importar arquivos de pastas:

```
ref1 <- pasta.nomeDoArquivo
ref2 <- pasta.subPasta.nomeDoArquivo
```

Para importar todos os arquivos de uma pasta, use a sintaxe de "ponto-sublinhado":

```
lib <- pasta._
```

Observe: apenas os arquivos imediatamente dentro da pasta são incluídos usando a sintaxe acima. Para incluir arquivos de subpastas, mencione-os explicitamente. Além disso, um arquivo `"_fat"` (ou arquivo `"_"`) dentro de uma pasta pode substituir o comportamento de importação de "ponto-sublinhado".

Acesso de elementos

Uma vez importado, acesse os elementos usando a sintaxe de ponto:

```
ref1.elemento1
```

Extração de elementos

Para extrair elementos específicos de uma importação nomeada ou para evitar adicionar o nome do módulo todas as vezes (por exemplo, `lib.foo`), use [atribuição por desestruturação](#):

```
{ foo, bar } = lib
```

Importação local

Para importar no escopo atual, use:

```
_ <- nomeDoArquivo
```

Importações locais, ao contrário das nomeadas, despejam o conteúdo do arquivo diretamente no escopo atual. Assim, um método importado pode ser invocado como `baz(arg)` em vez de `ref.baz(arg)`.

Embora as importações locais sejam bem apropriadas para importar [tipos](#) no escopo global, elas devem ser usadas com cautela ao importar conteúdos de biblioteca. O uso excessivo de importações locais pode levar a poluição do namespace, tornando mais desafiador seguir o código, porque fica menos aparente de onde vêm os métodos.

Importante: Importações nomeadas são resolvidas no escopo global, independentemente de onde forem declaradas. Isso significa que mesmo se você declarar uma importação nomeada dentro de uma função ou escopo local, ela será globalmente

acessível.

Caminhos literais

Com caminhos literais, você pode usar qualquer nome de arquivo ou extensão. No entanto, observe que essas importações não são avaliadas durante o [empacotamento](#), mas em tempo de execução. Aqui está um exemplo:

```
ref <- '_pasta/fonte-2.outro'
```

Você também pode usar [textos inteligentes](#) como caminhos literais:

```
base = 'pasta'  
arquivo = 'fonte.xyz'  
ref <- '{base}/{arquivo}'
```

Como o FatScript também aceita [sintaxe semelhante a JSON](#), você pode até mesmo carregar um arquivo JSON diretamente como uma importação:

```
json <- 'sample/data.json'
```

Lembre-se de que caminhos literais podem tornar seu código mais complexo e essas importações só podem ser resolvidas dinamicamente, então use-os com moderação.

Política de importação

A política "importar apenas uma vez" do FatScript utiliza um sistema de flags para evitar importações redundantes de arquivos. Se uma importação para um caminho já importado for encontrada, a declaração de importação é silenciosamente ignorada.

No entanto, se uma importação local for usada dentro do corpo de um método, a importação é executada todas as vezes que o método é invocado.

Entradas

Entradas

As entradas são pares chave-valor que existem no escopo onde são declaradas.

Nomeação

Os nomes das entradas (chaves) **não** podem começar com uma letra maiúscula, que é a distinção com relação aos [tipos](#). Os identificadores são sensíveis a maiúsculas e minúsculas, portanto "batatasfritas" e "batatasFritas" seriam considerados entradas diferentes.

A convenção recomendada é usar `camelCase` para as entradas.

você pode usar um nome arbitrário como chave usando [nomeação dinâmica](#)

Declaração e atribuição

Em FatScript, você pode declarar entradas simplesmente atribuindo um valor:

```
isOnline: Boolean = true
age: Number      = 25
name: Text       = 'João'
```

Os tipos também podem ser inferidos a partir da atribuição:

```
isOnline = true    # Boolean
age      = 25      # Number
name     = 'João'  # Text
```

Entradas imutáveis

Quando uma entrada é declarada em FatScript, ela é imutável por padrão. Isso significa que, uma vez que você atribui um valor a ela, não pode alterá-lo:

```
fruta = 'banana'
fruta = 'abacate' # gera AssignError (erro de atribuição) porque fruta é imutável
```

As entradas imutáveis são úteis quando você quer garantir que um valor permaneça constante durante a execução do programa.

Exceção à regra

Vale ressaltar que a imutabilidade se aplica apenas à própria entrada, mas não ao seu conteúdo quando se trata de um escopo.

No FatScript, surge um paradoxo, já que todas as seguintes afirmações são verdadeiras:

- Uma entrada imutável não pode ser alterada depois de definida.
- Um escopo pode sempre aceitar novas entradas.
- Uma entrada pode armazenar um escopo.

Veja:

```
s = { a = 1, b = 2 }
s.c = 3 # mesmo que s seja imutável, ele aceita o novo valor de c
s      # agora { a = 1, b = 2, c = 3 }
```

isso não é um bug, é um recurso

Isso ocorre porque os escopos são passados por referência e podem sempre ser "modificados" pela adição de novas entradas, mesmo que a entrada que o contém seja imutável. Então, neste caso, é o mesmo, mas não é exatamente o mesmo.

Por outro lado, [listas](#) declaradas como imutáveis se comportam de forma mais consistente com a regra, pois novas entradas não podem ser acrescentadas a elas.

Entradas mutáveis

Sim, você pode declarar entradas mutáveis, também conhecidas como variáveis. Para declarar uma entrada mutável, use o caractere til ~:

```
~ fruta = 'banana'
fruta = 'abacate' # ok
```

Observe que mesmo uma entrada mutável não pode mudar imediatamente seu tipo, a menos que seja apagada do escopo. Para apagar uma entrada, atribua `null` a ela e, em seguida, redeclare-a com um novo tipo. Mudar tipos é desencorajado pela sintaxe e não é recomendado, mas é possível:

```
~ color = 32 # cria a entrada color como um número mutável
color = 'blue' # gera um TypeError porque color é um número
color = null # a entrada é apagada
color = 'blue' # redefine color com um tipo diferente (Texto)
```

você deve declarar a entrada como mutável novamente usando o til ~ ao redefini-la após a exclusão se quiser que o próximo valor seja mutável

Entradas dinâmicas

Você pode criar entradas com nomes dinâmicos usando colchetes [ref]:

```
ref = 'pipoca' # texto será o nome da entrada

opcoes = { [ref] = 'é saborosa' }

opcoes.[ref] # sintaxe dinâmica: 'é saborosa', com acesso de leitura e gravação
opcoes(ref) # sintaxe de obtenção: 'é saborosa', mas o valor é somente leitura
opcoes.pipoca # sintaxe de ponto: 'é saborosa', mas deve seguir a nomenclatura
```

todas as declarações dinâmicas são entradas mutáveis

Essa funcionalidade permite definir dinamicamente os nomes dentro de um escopo e criar entradas com nomes que, de outra forma, não seriam aceitos pelo FatScript.

As entradas dinâmicas também podem usar referências numéricas, mas a referência é convertida em texto automaticamente, e.g.:

```
[ 5 ] = 'texto armazenado na entrada 5'
$self.[ '5' ] # devolve 'texto armazenado na entrada 5'
$self.[ 5 ] # devolve 'texto armazenado na entrada 5'
```

em um contexto diferente, não seguido por uma atribuição = ou precedido por notação de ponto ., a sintaxe dinâmica será interpretada como uma declaração de [lista](#)

Entradas especiais

Entradas com nomes que começam com o sublinhado _ são completamente livres e dinâmicas, não requerem til ~ e também podem mudar de tipo sem a necessidade de apagamento, como variáveis no JavaScript ou Python.

Atribuição por desestruturação

Você pode copiar os valores de um escopo em outro escopo assim:

```
_ <- fat.math
distancia = (posicao: Scope): Number -> {
  { x, y } = posicao # atribuição por desestruturação no escopo do método
  sqrt(x ** 2 + y ** 2) # calcula a distância entre a origem e (x, y)
}
distancia({ x = 3, y = 5 }) # 5.83095189485
```

Você também pode usar a atribuição por desestruturação para expor um determinado método ou propriedade de uma [importação nomeada](#):

```
console <- fat.console
{ log } = console
log('Olá Mundo')
```

usando essa sintaxe com importações, você pode escolher trazer para o escopo atual apenas os elementos da biblioteca que você está interessado em usar, evitando assim a poluição do namespace com nomes que não teriam uso ou poderiam entrar em conflito com os de sua própria autoria

Sintaxe semelhante a JSON

O FatScript também suporta sintaxe semelhante a JSON para declarar entradas:

```
"nada": null,                # entrada Void - comportamento distinto, veja abaixo
"estaOnline": true,         # entrada Boolean
"idade": 25,                # entrada Number
"nome": "João",             # entrada Text
"tags": [ "a", "b" ],      # entrada List
"opcoes": { "prop": "outra" } # entrada Scope
```

Embora possa parecer que [declarar "nada"](#) cria um valor "nada" de `null`, é importante observar que a "entrada resultante" na verdade não existe no escopo. Quando você tenta acessar esse "nada", o FatScript retorna `null`, mas se você tentar mapear o escopo, o nome dessa entrada estará faltando, pois nunca foi realmente criado.

É importante observar que as declarações semelhantes a JSON sempre criam entradas imutáveis, portanto você não pode adicionar o caractere `~` para torná-las mutáveis.

Tipos

Tipos

Os tipos são usados no FatScript para combinar dados e comportamentos, atuando como modelos para a criação de novas réplicas (instâncias).

Nomeação

Os nomes de tipo são sensíveis a maiúsculas e minúsculas, devendo começar com uma letra maiúscula.

A convenção recomendada para identificadores de tipo é `PascalCase`.

Tipos Nativos

O FatScript fornece vários tipos nativos:

- [Any](#) - qualquer coisa
- [Void](#) - nada
- [Boolean](#) - primitivo
- [Number](#) - primitivo
- [Text](#) - primitivo
- [Method](#) - função ou lambda
- [List](#) - como uma matriz ou pilha
- [Scope](#) - como um objeto ou dicionário
- [Error](#) - sim, para erros

No entanto, é necessário importar o [pacote type](#) para acessar os membros de protótipo de cada tipo.

Tipos Personalizados

Além de usar os tipos fornecidos pela linguagem ou por uma biblioteca externa, você também pode criar seus próprios tipos ou estender os existentes com novos comportamentos.

Declaração

Para definir um tipo personalizado no FatScript, você pode usar uma simples declaração de atribuição. A definição de tipo pode ser envolvida em parênteses ou chaves. Ambas as sintaxes são válidas e têm o mesmo efeito. Você também pode opcionalmente definir valores padrão para as propriedades do tipo, como mostrado nos seguintes exemplos:

```
# Definição de tipo usando chaves
Carro = { km: Number, cor: Text }
```

```
# Definição de tipo usando parênteses com valores padrão
Carro = (km = 0, cor = 'branco')
```

Unicidade Global

Embora a definição de tipo seja armazenada no [escopo](#) onde é declarada, o nome do tipo deve ser único no seu programa. Se você tentar definir um tipo com o mesmo nome de um já existente, mesmo em um escopo diferente, um `AssignError` será gerado, a menos que a definição seja idêntica, caso em que ela será ignorada.

Uso

Para criar instâncias de um tipo personalizado, chame o nome do tipo como se fosse um [método](#), opcionalmente passando valores para as propriedades:

```
# Uso do tipo com padrões
carro = Carro()
# saída: { km: Number = 0, cor: Text = 'branco' }
```

```
# Uso do tipo, definindo uma das propriedades
```

Tipos

```
carroVermelho = Carro(cor = 'vermelho')
# saída: { km: Number = 0, cor: Text = 'vermelho' }
```

```
# Uso do tipo, totalmente qualificado
carroVelho1 = Carro(cor = 'azul', km = 38000)
# substitui ambos os valores
```

```
# Uso do tipo, args usando sequência de props
carroVelho2 = Carro(41000, 'verde')
# substitui valores usando a ordem da definição do tipo
```

Por padrão, os tipos personalizados retornam um escopo de suas propriedades. No entanto, se você definir um método `apply`, o tipo poderá retornar um valor diferente. Por exemplo, aqui está um tipo personalizado `Soma` com um método `apply` que retorna a soma de suas propriedades `a` e `b`:

```
Soma = (a: Number, b: Number, apply = -> a + b)
Soma(1, 2) # saída: 3
```

observe que os métodos `apply` têm acesso direto às propriedades da instância

Neste exemplo, o tipo base de saída do `apply` é um número, não um escopo. Isso também significa que as propriedades originais do tipo personalizado são perdidas durante a instanciação e não podem ser acessadas novamente.

Membros do protótipo

Esses são um tipo especial de método, armazenados dentro da definição do tipo:

```
TipoComMembrosDePrototipo = {
~ a: Number
~ b: Number

setA = (novoA: Number) -> $self.a = novoA
setB = (novoB: Number) -> $self.b = novoB
soma = (): Number -> $self.a + $self.b
}
```

Neste exemplo, `setA`, `setB` e `soma` são membros do protótipo. Observe que precisamos usar `$self`, que é um [comando embutido](#) que fornece uma referência ao escopo da própria instância (ou método), para que nós pudéssemos ganhar acesso às propriedades.

Checando tipos

Se você não sabe qual é o tipo de uma entrada, pode simplesmente verificar comparando com um nome de tipo:

```
lugar = 'restaurante'
lugar == Number # false
lugar == Text   # true
```

alternativamente, use o método `typeof` da [biblioteca sdk](#) para extrair o nome do tipo

Alias de tipo

No FatScript, você pode criar subtipos atribuindo um nome diferente a um tipo existente. Isso significa que o novo tipo herdará todas as propriedades do tipo base. Aqui está um exemplo:

```
_ <- fat.type.Text
Id = Text # cria um alias
```

Observe que os aliases de tipo são hierárquicos e podem ser usados para classificar valores enquanto ainda herdam o mesmo comportamento. No entanto, embora o alias seja considerado igual ao tipo base, as instâncias do novo tipo não são consideradas iguais ao tipo base.

Para verificar se um valor é uma instância de um alias de tipo ou do tipo base, você pode usar o operador de comparação de menor-ou-igual `<=`. Isso permite que você aceite qualquer tipo na cadeia de aliases, até o tipo base. Aqui está um exemplo:

```
Id == Text # verdadeiro, já que Id é um alias de Text
x = Id(123) # id: Id = '123'
x == Text  # falso, no entanto x é do tipo Id, não Text
```

Tipos

```
x == Id      # verdadeiro, como o esperado x é do tipo Id
x <= Text   # verdadeiro, já que x é do tipo Id, que é um alias de Text
```

Essa funcionalidade permite uma validação refinada em tipos específicos, mantendo a flexibilidade de usar diferentes aliases para o mesmo tipo subjacente.

limitação: não é possível criar alias para `Any`, `Void`, `List` ou `Method`

Restrições de tipo

No FatScript, você pode declarar restrições de tipo para argumentos de método. Quando um método é chamado, o argumento é verificado automaticamente em relação à restrição de tipo. Se o argumento não for do tipo esperado ou um de seus subtipos, um `TypeError` é gerado.

Se a restrição de tipo for um tipo base, qualquer subtipo desse tipo também será aceito como argumento. No entanto, se a restrição de tipo for um subtipo, somente argumentos que correspondam ao subtipo serão aceitos. Aqui está um exemplo:

```
generalista = (x: Text) -> x
restritivo  = (x: Id)  -> x
```

Neste exemplo, o método `generalista` aceita argumentos `Text` e `Id`, porque `Id` é um subtipo de `Text`. O método `restritivo` aceita apenas argumentos `Id` e não `Text`, porque `Id` é um subtipo de `Text`, mas não o contrário.

É importante enfatizar que os tipos personalizados são derivados de `Scope`. Nesse contexto, `Scope` seria o tipo `generalista` para, por exemplo, o tipo personalizado `Carro`.

Inclusões de tipo (avanzado)

Ao definir um tipo, você pode adicionar os recursos de um tipo existente simplesmente mencionando-o na definição de tipo. Isso é chamado de inclusão de tipo.

Por exemplo, para criar um novo tipo `CarroAlugado` com as propriedades de `Carro` e uma propriedade adicional `preco`, você pode escrever:

```
CarroAlugado = {
  # Inclusões
  Carro

  # Propriedade adicional
  preco: Number
}
```

```
CarroAlugado(50) # { cor: Text = 'branco', km: Number = 0, preco: Number = 50 }
```

Se uma propriedade não estiver definida no novo tipo, ela herdará o valor padrão do tipo incluído. No exemplo acima, as propriedades `cor` e `km` do `Carro` estão presentes no `CarroAlugado`, com seus valores padrão.

Herdando métodos de protótipo

Suponha que continuemos a partir do exemplo anterior do tipo `TipoComMembrosDePrototipo` que tem duas propriedades `a` e `b`, e três métodos de protótipo `setA`, `setB` e `soma`. Para criar um novo tipo `ComMaisMembros` que adiciona uma propriedade `c`, um método `setC` e substitui o método `soma`, você pode escrever:

```
ComMaisMembros = {
  # Inclusões
  TipoComMembrosDePrototipo

  # Propriedades (argumentos da instância)
  ~ a: Number
  ~ b: Number
  ~ c: Number

  # Membros de protótipo (métodos)
  setC = (novoC: Number) -> $self.c = novoC
  soma = (): Number      -> $self.a + $self.b + $self.c
}
```

redeclarando as propriedades permite que o novo tipo também aceite argumentos no momento da instanciação, por exemplo: `ComMaisMembros(1, 2, 3)` define `a`, `b` e `c`

Ao criar uma nova instância de `ComMaisMembros`, todos os quatro métodos de protótipo `setA`, `setB`, `setC` e `soma` estarão disponíveis.

Observe que se houver uma redefinição de uma propriedade ou método no novo tipo, a nova definição terá precedência.

Conversão de tipos

Em `FatScript`, o símbolo `*` funciona como um operador de conversão de tipo, permitindo que você converta um tipo de dado em outro. Essa funcionalidade é especialmente útil quando você precisa especificar explicitamente o tipo ou realizar conversões entre tipos compatíveis, por exemplo:

```
time.format(Epoch * 1688257765448) # converte o número para Unix Epoch
```

Aceitação flexível de tipos

`FatScript` oferece flexibilidade na aceitação de tipos implementando um sistema baseado na inclusão de tipos. Isso cria tipos inter-relacionados que podem ser usados de forma intercambiável em um método ou como itens de uma `Lista`.

Quando você define um tipo, é possível incorporar um ou mais tipos adicionais dentro dessa definição. Por exemplo, os tipos `A`, `B`, e `C`. Se os tipos `B` e `C` incluem o tipo `A` em suas definições, eles são vistos como compartilhando o mesmo conjunto de características derivadas de `A`. Isso significa que `B` e `C` são considerados tipos irmãos sob o guarda-chuva de `A`.

Este sistema permite que um método que foi projetado para aceitar um objeto do tipo `B` também seja capaz de aceitar um objeto do tipo `C`, e vice-versa. Isso ocorre pelo fato de que ambos os tipos `B` e `C` compartilham uma base comum no tipo `A`.

Aqui está como isso parece no código:

```
A = ( )
B = (A, b = true)
C = (A, c = true)

# o método1 aceita tanto B quanto C, porque ambos incluem A
method1 = (a: A) -> 'valid'

# o método2 aceita C, já que B e C incluem o mesmo conjunto de tipos
# (tornando-os tipos irmãos)
method2 = (x: B) -> 'valid'

# essa lógica também se aplica a tipos de Lista, como visto com mixedList
mixedList: List/A = [ B(), C() ]
```

a flexibilidade do tipo só é possível se o tipo de dados é baseado em `Scope`

Advertência

Você pode ter que verificar explicitamente o tipo, por exemplo, `x == B` dentro do corpo do método se você quiser lidar apenas com `B`, mas não com `C` em seu método. Ou você pode criar um alias, por exemplo, `D = A` e usar `C = (D, c = true)` como inclusão de tipo para evitar completamente o comportamento flexível.

Tipos compostos

No `FatScript`, tipos compostos permitem que você defina estruturas de dados complexas compostas por tipos mais simples. Eles são representados usando barras `/` para separar os tipos na definição do tipo composto.

Vamos ver alguns exemplos e entender como os tipos compostos funcionam:

1. `ListOfNumbers = List/Number`, define um tipo composto `ListOfNumbers`, que é uma lista que só pode conter números.
2. `Matrix = List/List/Number`, define um tipo composto `Matrix`, que é uma lista de listas que só pode conter números.
3. `MethodReturningListOfNumbers = Method/ListOfNumbers`, define um tipo composto `MethodReturningListOfNumbers`, que é um método que retorna um `ListOfNumbers`.

4. `NumericScope = Scope/Number` define um tipo composto `NumericScope`, que é um escopo cujas entradas podem ser apenas do tipo número.

Veja também

- [Pacote type](#)

Any

Any

Um tipo virtual que engloba todos os tipos e nenhum tipo ao mesmo tempo.

Tipo padrão

Any é o tipo inferido e o tipo de retorno quando nenhum tipo é explicitamente anotado em um método. Por exemplo:

```
identity = _ -> _
```

é equivalente a:

```
identity = (_: Any): Any -> _
```

Usar Any, seja implicitamente ou explicitamente, desabilita a verificação de tipos para um argumento. A anotação explícita pode ser útil em casos em que você deseja deixar claro que está dando flexibilidade ao tipo de um parâmetro.

Ser muito liberal com Any pode tornar seu código menos previsível e mais difícil de manter. É geralmente recomendado ser mais específico com as anotações de tipo sempre que possível:

Exemplo de uso de Any que pode levar a problemas

```
console <- fat.console
```

```
doubleIt = (arg: Any): Void -> console.log(arg * 2)
```

```
doubleIt(2)    # imprime: '4'
doubleIt('a') # gera: Error: unsupported expression > Text <multiply> Number
```

Este exemplo mostra que, embora a anotação de tipo Any permita flexibilidade no tipo do argumento, também pode resultar em comportamento inesperado se um valor de um tipo inesperado for passado. Ao ser mais específico com a anotação de tipo, como Number, você pode tornar seu código mais previsível e autoexplicativo.

Exemplo de uso de uma anotação de tipo específica para maior previsibilidade

```
console <- fat.console
```

```
doubleIt = (num: Number): Void -> console.log(num * 2)
```

```
doubleIt(2)    # imprime: '4'
doubleIt('a') # gera: TypeError: type mismatch > num
```

Ao usar Number como a anotação de tipo, o método doubleIt agora é mais específico e só aceita argumentos do tipo Number.

Comparação

A única operação possível com Any é a comparação com ele, mas note que Any aceita todos os valores indistintamente, então não há uso prático para isso:

```
null      == Any # verdadeiro
true       == Any # verdadeiro
12345     == Any # verdadeiro
'abcd'    == Any # verdadeiro
[ 1, 2 ]  == Any # verdadeiro
{ a = 8 } == Any # verdadeiro
```

as comparações com Any não podem ser usadas para verificar a presença de um valor em um escopo, pois até mesmo null é aceito

Void

Void

Quando você olha para o 'Vazio', apenas 'nulo' pode ser visto.

Tem alguém aí fora?

Uma entrada é avaliada como `null` se não estiver definida no escopo atual.

Você pode comparar com `null` usando igualdade `==` ou desigualdade `!=`, como:

```
a == null # verdadeiro, se 'a' não estiver definida
0 != null # verdadeiro, porque 0 é um valor definido
```

Tenha em mente que você não pode declarar uma entrada sem valor no FatScript.

Embora você possa atribuir `null` a uma entrada, isso causa comportamentos diferentes, dependendo se a entrada já existe no escopo e se é mutável ou não:

- Se uma entrada ainda não foi declarada, atribuir `null` não tem efeito.
- Se já existe e é imutável, atribuir `null` gera um erro.
- Se já existe e é mutável, atribuir `null` remove a entrada.

Declaração de exclusão

Atribuir `null` a uma entrada mutável é o mesmo que excluir essa entrada do escopo. Se excluído, nada é lembrado sobre essa entrada no escopo, nem mesmo seu tipo original.

```
~ m = 4 # entrada de número mutável
m = null # exclui m do escopo
```

"valores" `null` são sempre mutáveis, pois na verdade nada é armazenado sobre eles e, portanto, são o único tipo de "valor" que pode fazer a transição de um estado mutável para um estado imutável quando "reatribuído"

Comparações

Você também pode usar `Void` para verificar o valor de uma entrada, como:

```
() == Void # verdadeiro
null == Void # verdadeiro
false == Void # falso
0 == Void # falso
'' == Void # falso
[] == Void # falso
{} == Void # falso
```

Observe que `Void` só aceita `null`.

Outra forma de vazio

Nulos também podem ser expressos como parênteses vazios `()` e são efetivamente idênticos, em termos de comportamento no código:

```
null == null # verdadeiro
() == null # verdadeiro
() == () # verdadeiro
```

Veja também

- [Extensões do protótipo Void](#)

Boolean

Boolean

Booleanos são muito primitivos, eles só podem ser 'verdadeiro' ou 'falso'.

Comparação

Além de igualdade `==` e desigualdade `!=`, os booleanos também aceitam os seguintes operadores:

& AND lógico

```
true & true == true
true & false == false
false & true == false
false & false == false
```

AND interrompe a expressão se o lado esquerdo for falso

| OR lógico

```
true | true == true
true | false == true
false | true == true
false | false == false
```

OR interrompe a expressão se o lado esquerdo for verdadeiro

% XOR lógico (OR exclusivo)

```
true % true == false
true % false == true
false % true == true
false % false == false
```

XOR sempre avalia ambos os lados da expressão

Operador Bang

!! converte qualquer tipo em booleano, assim:

- null -> false
- zero (número) -> false
- não-zero (número) -> true
- vazio (texto/lista/escopo) -> false
- não-vazio (texto/lista/escopo) -> true
- método -> true

fluxos condicionais (`=>`, `?`) converterão implicitamente o lado esquerdo em booleano

Veja também

- [Extensões do protótipo Boolean](#)
- [Controle de fluxo](#)

Number

Number

Um conceito matemático usado para contar, medir e fazer outras coisas de [matemáticas](#).

Declaração

O tipo `Number` é implementado como `double`. Veja como declarar um número:

```
a = 5           # declaração de número (imutável)
b: Number = 5  # mesmo efeito, com verificação de tipo
c: Number = a  # iniciando com o valor da entrada, também 5
d = 43.14      # com casas decimais
```

Para declarar uma entrada mutável, coloque o operador til antes:

```
~ a = 6 # entrada de número mutável
a += 1  # adiciona 1 a 'a', resultando em 7
```

Operações com números

Números aceitam várias operações:

- `==` igual
- `!=` diferente
- `+` soma
- `-` subtração
- `*` multiplicação
- `/` divisão
- `%` módulo
- `**` potência
- `<` menor
- `<=` menor ou igual
- `>` maior
- `>=` maior ou igual
- `&` AND lógico
- `|` OR lógico

Ressalvas

Para operações lógicas e controle de fluxo, lembre-se de que zero é considerado falso e um não-zero é considerado verdadeiro.

Para operadores de igualdade, embora `0` e `null` sejam avaliados como falsos, no FatScript eles não são iguais:

```
0 == null # falso
```

Precisão

Embora a precisão aritmética de um `IEEE 754 double` seja maior, o `fry` utiliza truques de arredondamento para melhorar a legibilidade humana ao imprimir sequências longas de nove ou zeros decimais como texto. Além disso, ele usa um `epsilon` de `1.0e-06` para comparações de 'igualdade' entre números.

Em 99,999% dos casos de uso, essa abordagem fornece tanto comparações mais convenientes quanto números mais naturais:

```
# Epsilon de igualdade
x = 1.0e-06
x: Number = 0.000001

# Diferenças menores são tratadas como o "mesmo" número pela comparação
x == 0.0000015
Boolean: true # a diferença de 0,0000005 é ignorada
```


Text

Text

Textos podem conter muitos caracteres e são às vezes chamados de strings.

Declaração

Entradas de texto são declaradas usando aspas:

```
a = 'hello world'      # declaração de texto inteligente
a = "hello world"     # declaração de texto bruto
a: Text = 'hello world' # inteligente, opcionalmente verboso
```

Manipulando texto

Concatenação

No FatScript, você pode concatenar, ou juntar, dois textos usando o operador +. Essa operação conecta os dois textos em um. Por exemplo:

```
x1 = 'ab' + 'cd' # Retorna 'abcd'
```

Subtração de texto

FatScript também suporta uma operação de subtração de texto usando o operador -. Essa operação remove uma substring especificada do texto. Por exemplo:

```
x2 = 'ab cd'
x2 - ' ' == 'abcd' # Retorna true
```

No exemplo acima, o caractere de espaço ' ' é removido do texto original 'ab cd', resultando em 'abcd'.

Seleção de texto

A seleção permite que você acesse partes específicas de um texto usando índices. No FatScript, você pode usar índices positivos ou negativos. Os índices positivos começam do início do texto (0 é o primeiro caractere), e os índices negativos começam do final do texto (-1 é o último caractere).

para uma explicação detalhada sobre o sistema de indexação no FatScript, consulte a seção sobre acesso e seleção de itens em [List](#)

Quando apenas um índice é passado para a função de seleção, um único caractere do texto é selecionado. Quando dois índices são passados para a função, um intervalo de caracteres do texto é selecionado. Essa seleção é inclusiva, o que significa que inclui os caracteres nos índices inicial e final.

Assim como com as listas, acessar itens que estão fora dos índices válidos irá gerar um erro. Para seleções, não são gerados erros ao acessar índices fora dos limites; em vez disso, um texto vazio é retornado.

```
x3 = 'exemplo'
x3(1) # 'x'
x3(2, 4) # 'emp'
x3(..2) # 'exe'
```

Caracteres especiais

Caracteres como aspas ' / " podem ser escapados com a barra invertida \.

```
'Rock\n\roll'
"Onde fica \"aqui\"?"
```

você só precisa escapar as aspas do mesmo tipo usadas como delimitador de texto

Outras sequências de escape suportadas são:

- backspace `\b`
- nova linha `\n`
- retorno de carro `\r`
- tabulação `\t`
- escape `\e`
- octeto em representação base-8 `\ooo`
- a própria barra invertida `\\`

Textos inteligentes

Quando declarado com aspas simples `'`, o modo inteligente é habilitado e a interpolação é realizada para qualquer código envolto em chaves `{...}`:

```
texto = 'mundo'
interpolado = 'olá {texto}' # resulta em 'olá mundo'
```

o template é processado em uma camada com acesso ao escopo atual

Observe que o uso de novas linhas ou outros textos inteligentes dentro do template de interpolação não é suportado, mas você pode fazer chamadas de método, se precisar compor o resultado com algo mais complexo.

Você pode evitar a interpolação escapando o colchete de abertura:

```
escapado = 'olá \{texto}' # resulta em 'olá {texto}'
```

Alternativamente, você pode evitar a interpolação usando textos brutos.

Textos brutos

Quando declarado com aspas duplas `"`, o modo de texto bruto é assumido e a interpolação é desativada.

Exemplo de modo inteligente vs. modo bruto:

```
'Sou inteligente: {interpolado}' # usando o valor do exemplo anterior
Sou inteligente: olá mundo      # substituição ocorre
```

```
"Sou bruto: {interpolado}" # colchetes são apenas caracteres comuns
Sou bruto: {interpolado}   # nenhuma interpolação ocorre
```

Operações com textos

- `==` igual
- `!=` diferente
- `+` soma (concatenar)
- `-` subtração (remove substring)
- `<` menor (alfanumérico)
- `<=` menor ou igual (alfanumérico)
- `>` maior (alfanumérico)
- `>=` maior ou igual (alfanumérico)
- `&` AND lógico (convertido para booleano)
- `|` OR lógico (convertido para booleano)

comparações são implementadas através da função [strcmp](#).

Codificação

FatScript é projetado para operar com textos codificados em UTF-8 ou ASCII. Essa escolha de design reconhece a prevalência desses sistemas de codificação e otimiza a linguagem para ampla compatibilidade.

UTF-8 é um sistema de codificação de vários bytes capaz de representar qualquer caractere no padrão Unicode. Este esquema de codificação de caracteres universais usa de 8 a 32 bits para representar um caractere, permitindo a representação de uma vasta gama de símbolos de diversas línguas e sistemas de escrita. Notavelmente, os primeiros 128 caracteres (0-127) do UTF-8 se alinham precisamente com o conjunto ASCII, tornando qualquer texto ASCII uma string válida codificada em UTF-8.

No FatScript, o tipo de dados `Text` é uma sequência de caracteres Unicode, inerentemente codificada em UTF-8, portanto, operações como `text.size`, `text(index)` e `text(1..4)` irão contar, acessar ou fatiar corretamente o texto,

independentemente da complexidade dos caracteres. Essas operações consideram um caractere UTF-8 multi-byte completo como uma única unidade, garantindo um comportamento correto e previsível.

Ao assumir a codificação UTF-8 para o texto, FatScript garante interoperabilidade perfeita com os padrões existentes, amplia sua aplicabilidade em várias línguas e scripts e melhora a experiência do usuário ao tratar textos como sequências logicamente contíguas de caracteres.

Veja também

- [Extensões do protótipo Text](#)

Method

Method

Métodos são receitas que podem receber argumentos para "preencher as lacunas em branco".

Definição

Um método é definido anonimamente com uma seta fina `->`, assim:

```
<argumentos> -> <receita>
```

Os argumentos podem ser omitidos se nenhum for necessário:

```
-> <receita> # aridade zero
```

Para registrar um método no escopo, atribua-o a um identificador:

```
<identificador> = <argumentos> -> <receita>
```

O número de argumentos na assinatura de um método é fixo e todos os argumentos são obrigatórios. No entanto, se você passar um argumento adicional para o método, ele pode ser acessado como [implícito](#). Outros serão ignorados.

Os argumentos são tratados como entradas imutáveis dentro do escopo de execução do método. Se você precisar de argumentos opcionais ou mutáveis, poderá passar um escopo como argumento, que pode agrupar vários "argumentos". Alternativamente, você pode usar um [tipo personalizado](#) com valores padrão.

Auto-retorno

FatScript usa o auto-retorno, ou seja, o último valor é retornado automaticamente:

```
resposta: Method = (aGrandeQuestao) -> {
  # TODO: explicar a vida, o universo e tudo mais
  42
}
```

```
resposta("6 x 7 = ?") # retorna: 42
```

Truque de chamada automática

Se um método é acessado dentro de um escopo com a sintaxe de ponto e não requer argumentos (aridade zero), ele será chamado "automaticamente":

```
foo = {
  bar = -> 'oi'
}
```

```
foo.bar() # retorna 'oi'
foo.bar   # também retorna 'oi' devido à chamada automática
```

Nota: esses métodos ainda podem ser referenciados sem acionar o recurso de chamada automática usando a sintaxe de obtenção:

```
foo('bar') # retorna uma referência ao método
```

Argumento implícito

Uma conveniência oferecida pelo FatScript é a possibilidade de fazer referência a um valor passado para o método sem precisar especificar um nome a ele explicitamente. Neste caso, o argumento implícito é então representado pelo sublinhado `_`.

Aqui está um exemplo que ilustra o uso de argumentos implícitos:

```
dobro = -> _ * 2
dobro(3) # saída: 6
```

Você pode usar argumentos implícitos sempre que precisar realizar uma operação simples em um único argumento sem atribuir um nome específico a ele.

Veja também

- [Extensões do protótipo Method](#)

List

List

Listas são coleções ordenadas de itens do mesmo tipo, acessados por índice.

Definição

As listas são definidas com colchetes [], como no exemplo a seguir:

```
lista: List/Text = [ 'maçã', 'pizza', 'pêra' ]
```

Listas não permitem a mistura de tipos. O tipo de uma lista é determinado pelo primeiro item adicionado a ela; consequentemente, as listas vazias não têm tipo.

As listas pulam posições vazias, então um item que avalia para `null` é ignorado:

```
a = 1
c = 3
[ a, b, c ] # retorna: [ 1, 3 ] (b é ignorado)
```

Acesso

Itens individuais

Os itens da lista podem ser acessados individualmente com chamada de índice baseado em zero:

```
lista(0) # 'maçã'
lista(2) # 'pêra'
```

Valores negativos indexam a lista de trás para frente, começando em -1 como o último item:

```
lista(-1) # 'pêra'
```

O acesso a itens que estão fora dos índices válidos gera um erro:

```

      0      1      2      > 2
Erro [ 'maçã', 'pizza', 'pêra' ] Erro
    < -3      -3      -2      -1
```

Seleções

Você pode passar um segundo argumento para realizar uma chamada de acesso como seleção do índice "início" até o índice "fim", avaliando os índices como inclusivos.

Os índices de início e fim funcionam exatamente da mesma maneira que ao acessar itens individuais; então, os valores negativos contam a partir do último item e podem ser regressivos. No entanto, ao usar seleções, nenhum erro é gerado ao acessar índices fora dos limites; em vez disso, uma lista vazia é retornada.

```
lista(0, 0) # [ 'maçã' ]
lista(4, 8) # []
lista(1, -1) # [ 'pizza', 'pêra' ]
```

O mesmo vale para a sintaxe de intervalo (..):

```
lista(0..0) # [ 'maçã' ]
lista(4..8) # []
lista(1..-1) # [ 'pizza', 'pêra' ]
```

No entanto, com intervalos, um índice pode ser deixado em branco e assume início a partir do primeiro item ou fim no último item:

```
lista(..1) # [ 'maçã', 'pizza' ]
lista(1..) # [ 'pizza', 'pêra' ]
```


Listas aninhadas

Uma matriz pode ser usada e acessada da seguinte maneira:

```
matriz = [
  [ 1, 2, 3 ]
  [ 4, 5, 6 ]
]
```

```
matriz(1)(0) # retorna 4 (1: segunda linha, em seguida, 0: primeiro índice)
```

para simplificar, o exemplo usa uma matriz 2D, mas poderia ser n-dimensional

Operações

- == igual
- != diferente
- + adição (efeito de concatenação)
- - subtração (efeito de diferença)
- & AND lógico
- | OR lógico

AND/OR lógicos avaliam listas vazias como `false`, caso contrário `true`

Adição de lista (concatenação)

A operação de adição de listas permite combinar duas listas em uma nova lista:

```
x = [ 1, 2, 2, 3 ]
y = [ 3, 3, 4, 4 ]
```

```
x + y # resultado: [ 1, 2, 2, 3, 3, 3, 4, 4 ]
```

Nesse caso, ao usar o operador de adição `+` para unir as listas `x` e `y`, os elementos de ambas as listas são combinados em uma única lista. A ordem dos elementos na lista resultante é determinada pela ordem em que as listas foram adicionadas.

não há remoção de elementos duplicados durante a concatenação

Concatenação rápida

Para melhor desempenho, você pode aproveitar o operador `+=`, por exemplo:

```
~ lista += [ valor ] # mais rápido
```

```
# mesmo efeito que
```

```
~ lista = []
```

```
lista = lista + [ valor ] # concatenação (mais lenta)
```

Outro detalhe do operador `+=`, que se aplica também a outros tipos, é a inicialização automática por omissão, onde caso a entrada ainda não tenha sido declarada anteriormente, atua como uma simples atribuição.

Subtração de lista (diferença)

A operação de subtração de listas, permite remover os elementos do segundo operando que estão presentes no primeiro operando, resultando em uma lista contendo apenas valores únicos:

```
x = [ 1, 2, 2, 3 ]
y = [ 3, 3, 4, 4 ]
```

```
x - y # resultado: [ 1, 2 ]
y - x # resultado: [ 4 ]
```

Nesse caso, ao subtrairmos a lista `y` da lista `x`, os elementos com valor 3 são removidos, já que estão presentes em ambas as listas. O resultado é a lista `[1, 2]`. Da mesma forma, ao subtrairmos a lista `x` da lista `y`, o único elemento restante é o valor 4.

apenas valores exatamente idênticos são removidos durante a subtração

Veja também

- [Extensões do protótipo List](#)
- [Mapeando uma lista](#)

Scope

Scope

Um escopo é como um dicionário com entradas dentro, onde as chaves possuem valores.

Definição

Escopos são definidos com chaves {}, como no exemplo a seguir:

```
meuEscopoBacana = {
  lugar = 'aqui'
  quando = 'agora'
}
```

Escopos armazenam entradas em ordem alfabética. Isso se torna evidente quando você [mapeia](#) um escopo.

Acesso

Existem três maneiras de acessar diretamente as entradas dentro de um escopo.

Sintaxe de ponto

```
meuEscopoBacana.lugar # retorna: 'aqui'
```

Sintaxe de obtenção

```
# assumindo que prop = 'lugar'
meuEscopoBacana(prop) # retorna: 'aqui'
```

De qualquer maneira, se a propriedade não estiver presente, `null` será retornado. E se o escopo externo não for encontrado no escopo, um erro será gerado.

Sintaxe de encadeamento opcional

Você pode encadear o escopo externo ausente usando o operador interrogação-ponto `?.`:

```
naoExistente?.propriedade # null
```

O encadeamento opcional não gera um erro quando o escopo externo é `null`.

Operações

- `==` igual
- `!=` diferente
- `+` adição (efeito de mesclagem)
- `-` subtração (efeito de diferença)
- `&` AND lógico
- `|` OR lógico

logical AND/OR evaluate empty scopes as `false`, otherwise `true`

AND/OR lógicos avaliam escopos vazios como `false`, caso contrário `true`

Adição de escopo (mesclagem)

O segundo operando funciona como se fosse um patch para o primeiro operando:

```
x = { a = 1, b = 3 }
y = { b = 2 }

x + y # { a = 1, b = 2 }
y + x # { a = 1, b = 3 }
```

valores do segundo operando substituem valores do primeiro operando

Subtração de escopo (diferença)

A operação de subtração, gera a remoção dos elementos do segundo operando que também estão presentes no primeiro operando:

```
x = { a = 1, b = 3 }  
y = { a = 1 }
```

```
x - y # { b = 3 }
```

apenas valores exatamente idênticos são removidos

Veja também

- [Entradas dinâmicas](#)
- [Extensões do protótipo Scope](#)
- [Mapeando um escopo](#)

Error

Error

Há grande sabedoria em esperar pelo inesperado também.

Subtipos padrão

Enquanto alguns erros genéricos, como problemas de sintaxe, importações inválidas, etc. são gerados com o tipo base `Error`, outros são [subtipados](#):

- `KeyError`: a chave (nome) não é encontrada no escopo
- `IndexError`: o índice está fora dos limites da lista/texto
- `CallError`: uma chamada é feita com argumentos insuficientes
- `TypeError`: inconsistência de tipo em chamada, retorno ou atribuição de método
- `AssignError`: atribuindo um novo valor a uma entrada imutável
- `ValueError`: tipo pode estar correto, mas conteúdo não é aceito

Comparações

Erros sempre avaliam como falso:

```
Error() ? 'é verdadeiro' : 'é falso' # é falso
```

Erros são comparáveis ao seu tipo:

```
Error() == Error # verdadeiro
```

leia também a sintaxe de [comparação de tipo](#)

Uma maneira ingênua de lidar com erros poderia ser:

```
_ <- fat.console
# lidando com o erro retornado
maybeFail() <= Error => log('um erro aconteceu')
_                    => log('sucesso')
```

isso só funciona se a [opção](#) `-e` / `continuar` em caso de erro estiver definida

Embora a abordagem ingênua possa funcionar, é difícil saber onde os erros surgirão. Portanto, uma maneira mais adequada de lidar com erros é definindo um manipulador de erro usando o método `trapWith` encontrado na [biblioteca failure](#).

Veja também

- [Biblioteca failure](#)
- [Extensões do protótipo Error](#)

Controle de fluxo

Controle de fluxo

Avance em um fluxo contínuo de decisões que devem ser tomadas.

Fallback

As operações padrão ou de coalescência nula são definidas com dois pontos de interrogação ?? e funcionam da seguinte maneira:

```
<expressãoTalvezNulaOuFalha> ?? <valorFallback>
```

Caso o lado esquerdo não seja null nem Error, então ele é usado; caso contrário, o valor de fallback é retornado.

If

Declarações If são definidas com um ponto de interrogação ?, como abaixo:

```
<condição> ? <resposta>
```

como não há alternativa, null é retornado se a condição não for atendida

If-Else

Declarações If-Else são definidas com um ponto de interrogação ? seguido de dois pontos :, como abaixo:

```
<condição> ? <resposta> : <alternativa>
```

Para usar declarações If-Else multilinhas, envolva a resposta em chaves {...} assim:

```
<condição> ? {
  <resposta>
} : {
  <alternativa>
}
```

Cases

Cases são definidos com a seta espessa => e são automaticamente encadeados, criando uma sintaxe intuitiva e simplificada, semelhante a uma declaração switch, sem a possibilidade de queda. Isso permite que condições não relacionadas sejam misturadas, resultando em uma estrutura if-else-if-else mais concisa:

```
<condição1> => <respostaPara1>
<condição2> => <respostaPara2>
<condição3> => <respostaPara3>
...
```

Exemplo:

```
escolha = (x) -> {
  x == 1 => 'a'
  x == 2 => 'b'
  x == 3 => 'c'
}
```

```
escolha(2) # 'b'
escolha(8) # null
```

Para fornecer um valor padrão para seu método, você pode adicionar um caso pega-tudo usando um sublinhado _ no final da sequência:

```
escolha = (x) -> {
  x == 1 => 'a'
```

Controle de fluxo

```
x == 2 => 'b'  
x == 3 => 'c'  
_      => 'd'  
}
```

```
escolha(2) # 'b'  
escolha(8) # 'd'
```

Para cenários mais complexos, você pode usar blocos como resultados para cada caso:

```
...  
condição => {  
  # faça algo  
  'foo'  
}  
_ => {  
  # faça outra coisa  
  'bar'  
}  
...
```

Cases devem terminar em um caso pega-tudo `_` ou final do bloco. O uso mais efetivo de Cases é dentro de métodos na parte inferior do corpo do método.

Embora seja possível adicionar Cases aninhados, é melhor evitar construções excessivamente complexas. Isso torna o código mais difícil de seguir e provavelmente perde o objetivo de usar esse recurso.

Pode ser mais apropriado extrair essa lógica para um método separado. O FatScript incentiva os desenvolvedores a dividir a lógica em métodos distintos, ajudando a evitar código spaghetti.

Loops

Loops

Repetir, repetir, repetir, repetir, repetir...

Sintaxe base

Todos os loops são criados com o sinal de arroba @, por exemplo:

```
<expressão> @ <corpoDoLoop>
```

Loop tipo "enquanto"

O corpo do loop irá executar enquanto a expressão avaliar para:

- verdadeiro
- número não zero
- texto não vazio

A execução irá terminar quando a expressão avaliar para:

- falso
- nulo
- número zero
- texto vazio
- erro

Por exemplo, este loop imprime números de 0 a 3:

```
_ <- fat.console
~ i = 0
(i < 4) @ {
  log(i)
  i += 1
}
```

Sintaxe de mapeamento

Você pode mapear intervalos, listas e escopos com um mapeador, assim:

```
<intervalo|coleção> @ <mapeador>
```

Uma nova lista é gerada com base nos valores de retorno do mapeador.

Mapeando um intervalo

Utilizando o operador de intervalo `..` o mapeador receberá um número como entrada sequencialmente do limite esquerdo até o limite direito:

```
4..0 @ num -> num + 1 # retorna [ 5, 4, 3, 2, 1 ]
```

a sintaxe de intervalo é inclusiva em ambos os lados, por exemplo, `0..2` retorna 0, 1, 2.

Há também o operador de intervalo semiaberto `..<`, exclusivo no lado direito.

ressalva: o intervalo semiaberto não funciona com direção inversa, sempre precisa ser do mínimo para máximo

Mapeando uma lista

O mapeador receberá os itens em ordem (da esquerda para a direita):

```
[ 3, 1, 2 ] @ item -> item + 1 # retorna [ 4, 2, 3 ]
```


Mapeando um escopo

O mapeador receberá os nomes (chaves) das entradas armazenadas no escopo em ordem alfabética:

```
{ c = 3, a = 1, b = 2 } @ chave -> chave # retorna [ 'a', 'b', 'c' ]
```

nos exemplos, usamos literais de lista e escopo, mas uma entrada ou chamada que avalia para uma lista ou um escopo terá o mesmo efeito

Você pode acessar as entradas de um escopo referindo-se a ele pelo nome, mas neste caso precisa que ele esteja definido no escopo externo, por exemplo:

```
meuEscopo = { c = 3, a = 1, b = 2 }  
meuEscopo @ chave -> meuEscopo(chave) # retorna [ 1, 2, 3 ]
```

O FatScript utiliza um recurso de caching inteligente que faz com que esta sintaxe não gere um esforço adicional para buscar o elemento da vez no escopo durante o mapeamento.

Bibliotecas

Bibliotecas

Vamos falar sobre os doces recheios embutidos no FatScript: as bibliotecas!

Bibliotecas padrão

Essenciais

Estas são as bibliotecas fundamentais que você espera que estejam disponíveis em uma linguagem de programação, fornecendo funcionalidades essenciais:

- [async](#) - Trabalhadores e tarefas assíncronas
- [color](#) - Códigos de cores ANSI para console
- [console](#) - Operações de entrada e saída do console
- [curses](#) - Interface de usuário baseada em terminal
- [failure](#) - Tratamento de erros e gerenciamento de exceções
- [file](#) - Operações de entrada e saída de arquivos
- [http](#) - Framework de manipulação HTTP
- [math](#) - Operações e funções matemáticas
- [sdk](#) - Utilitários do kit de desenvolvimento de software da Fry
- [system](#) - Operações e informações no nível do sistema
- [time](#) - Manipulação de data e hora
- [zCode](#) - Métodos de codificação de dados, hash e uuid

Pacote de tipos

[Este pacote](#) estende os recursos dos [tipos nativos](#) do FatScript:

- [Void](#)
- [Boolean](#)
- [Number](#)
- [Text](#)
- [Method](#)
- [List](#)
- [Scope](#)
- [Error](#)

Pacote Extra

[Estas utilidades](#) são implementadas em FatScript puro:

- [csv](#) - Codificador e decodificador rudimentar de CSV
- [Date](#) - Gerenciamento de calendário e datas
- [Duration](#) - Construtor de duração em milissegundos
- [elapsed](#) - Calculadora de tempo decorrido
- [HashMap](#) - Armazenamento rápido de chave-valor
- [hex](#) - Codificador e decodificador hexadecimal
- [json](#) - Codificação e armazenamento de dados JSON
- [Logger](#) - Suporte ao registro de logs
- [mathex](#) - Biblioteca matemática estendida
- [Memo](#) - Classe de utilidade de memoização genérica
- [regex](#) - Padrões comuns de expressões regulares
- [Sound](#) - Interface de reprodução de som
- [util](#) - Outras utilidades aleatórias
- [xml](#) - Analisador e gerador de XML simplificado

Atalho de importação

Se você deseja torná-los todos disponíveis de uma vez, pode simplesmente fazer o seguinte, e todas essas coisas boas estarão disponíveis para o seu código:

```
_ <- fat._
```

Embora esse recurso possa ser conveniente ao experimentar no REPL, esteja ciente de que ele traz todas as constantes e nomes de método da biblioteca, potencialmente poluindo seu namespace global.

Além disso, a importação de tudo antecipadamente pode adicionar uma sobrecarga desnecessária ao tempo de inicialização do seu programa, mesmo que você precise usar apenas alguns métodos.

Como boa prática, considere importar apenas os módulos específicos de que você precisa, com [importações nomeadas](#). Dessa forma, você pode manter seu código limpo e conciso, minimizando o risco de conflitos de nome ou problemas de desempenho.

Hacking e mais

Sob o capô, as bibliotecas são construídas usando comandos embutidos. Para obter uma compreensão mais profunda e explorar o funcionamento interno do interpretador, mergulhe [neste tópico mais avançado](#).

async

async

Trabalhadores e tarefas assíncronas

Importação

```
_ <- fat.async
```

Tipos

A biblioteca `async` introduz dois tipos: `Worker` e `Task`

Worker

O `Worker` é um simples invólucro para uma operação assíncrona.

Construtor

Nome	Assinatura	Breve descrição
<code>Worker</code>	<code>(task: Method)</code>	Cria um <code>Worker</code> em modo de espera

Membros do protótipo

Nome	Assinatura	Breve descrição
<code>start</code>	<code>() : Worker</code>	Inicia a tarefa
<code>cancel</code>	<code>() : Void</code>	Cancela a tarefa
<code>await</code>	<code>() : Worker</code>	Aguarda pela conclusão da tarefa
<code>isDone</code>	<code>() : Boolean</code>	Verifica se a tarefa foi concluída
<code>hasStarted</code>	<code>Boolean</code>	Definido pelo método <code>start</code>
<code>hasAwaited</code>	<code>Boolean</code>	Definido pelo método <code>await</code>
<code>isCanceled</code>	<code>Boolean</code>	Definido pelo método <code>cancel</code>
<code>result</code>	<code>Any</code>	Definido pelo método <code>await</code>

Task

Uma `Task` representa uma operação assíncrona com tempo limitado. Ela usa dois `Workers`, um para a tarefa em si e outro para o temporizador de tempo limite.

Construtor

Nome	Assinatura	Breve descrição
<code>Task</code>	<code>(task: Method, wait: Number)</code>	Cria uma <code>Task</code> em modo de espera

O construtor `Task` leva dois argumentos:

- **task**: O método a ser executado de forma assíncrona (o método pode não aceitar argumentos diretamente, mas você pode adicionar esses usando duas setas na definição `-> ->`).
- **wait** (opcional): O tempo limite em milissegundos. Se a tarefa não terminar dentro desse tempo, ela será cancelada. O padrão é 40.000ms (40 segundos).

Membros do protótipo

Nome	Assinatura	Breve descrição
<code>start</code>	<code>() : Task</code>	Inicia a tarefa e seu temporizador

Nome	Assinatura	Breve descrição
cancel	(): Void	Cancela a tarefa e seu temporizador
await	(): Task	Aguarda pela conclusão da tarefa ou timeout
isDone	Boolean	Definido true se a tarefa foi concluída
hasTimedOut	Boolean	Definido true se a tarefa excedeu o tempo
result	Any	Definido pelo método await

Método avulso

Nome	Assinatura	Breve descrição
selfCancel	(): *	Encerra a execução da thread

Notas de uso

As instâncias de `Worker` são mapeadas para threads do sistema em uma base um-para-um e são executadas conforme o agendamento do sistema. Isso implica que sua execução pode nem sempre ser imediata. Para aguardar o resultado de um `Worker` ou `Task`, use o método `await`.

Exemplos

```

_ <- fat.async
math <- fat.math
time <- fat.time

# Define uma tarefa lenta
slowTask = -> {
  waitTime = math.random * 5000 # Aguarda até 5 segundos
  time.wait(waitTime)
  waitTime
}

# Inicia a tarefa como Worker
worker = Worker(slowTask).start

# Obtém o resultado do worker
result1 = worker.await.result # bloqueia até a tarefa ser concluída

# Inicia uma tarefa com timeout
task = Task(slowTask, 3000).start

# Obtém o resultado da tarefa
result2 = task.await.result # bloqueia até a tarefa ser concluída ou ocorrer timeout

o método await da Task gera um AsyncError se a tarefa exceder o tempo antes da conclusão

```

Veja também

- [Biblioteca Time](#)
- [Pacote Type](#)

color

color

Códigos de cores ANSI para console

Importação

```
_ <- fat.color
```

Constantes

- black, 0
- red, 1
- green, 2
- yellow, 3
- blue, 4
- magenta, 5
- cyan, 6
- white, 7
- bright.black, 8
- bright.red, 9
- bright.green, 10
- bright.yellow, 11
- bright.blue, 12
- bright.magenta, 13
- bright.cyan, 14
- bright.white, 15

Métodos

Nome	Assinatura	Breve descrição	
detectDepth	()	Number	Obter suporte de cor do console
to8	(xr: Any, g: Number = \emptyset , b: Number = \emptyset)		Converter RGB para 8-cores
to16	(xr: Any, g: Number = \emptyset , b: Number = \emptyset)		Converter RGB para 16-cores
to256	(xr: Any, g: Number = \emptyset , b: Number = \emptyset)		Converter RGB para 256-cores

Notas de Uso

to8, to16 e to256

O parâmetro `xr` pode ser um texto opcional que representa a cor no formato HTML. Por exemplo, pode ser fornecido como `'fae830'` ou `'#fae830'` (amarelo):

```
color <- fat.console
console <- fat.console
```

```
console.log('hey', color.to16('fae830'))
console.log('hey', color.to256('fae830'))
```

No entanto, se `xr` for um número entre 0 e 255 representando `r`, então os parâmetros `g` e `b` serão necessários:

```
console.log('hey', color.to256(250, 232, 48)) // mesmo resultado
```

estes métodos podem produzir aproximações da cor original nas profundidades 8, 16 ou 256 e não a cor verdadeira exata

Veja também

- [Biblioteca console](#)
- [Biblioteca curses](#)

color

- [256 Cores](#)
-

console

console

Operações de entrada e saída do console

Importação

```
_ <- fat.console
```

Métodos

Nome	Assinatura	Breve descrição
log	(msg: Any, fg: Number = 0, bg: Number = 0): Void	Imprime msg para stdout, com quebra de linha
print	(msg: Any, fg: Number = 0, bg: Number = 0): Void	Imprime msg para stdout, sem quebra de linha
stderr	(msg: Any, fg: Number = 0, bg: Number = 0): Void	Imprime msg para stderr, com quebra de linha
input	(msg: Any, mode: Text = null): Text	Imprime msg e retornar entrada de stdin
flush	(): Void	Esvazia buffer de saída padrão
cls	(): Void	Limpa stdout usando códigos de escape ANSI
moveTo	(x: Number, y: Number): Void	Move o cursor usando códigos de escape ANSI
isTty	(): Boolean	Verifica se stdout é um terminal
showProgress	(label: Text, fraction: Number): Void	Renderiza barra de progresso, fração 0 a 1

os métodos de saída desta biblioteca garantem segurança de threads em cenários assíncronos

Notas de uso

saída

Por padrão, `stdout` e `stderr` imprimem no console. Os parâmetros de cor de primeiro plano (`fg`) e cor de plano de fundo (`bg`) são opcionais.

as cores são automaticamente suprimidas se o buffer de saída não for um TTY

input

O parâmetro opcional `mode` aceita os seguintes valores:

- 'plain', entrada simples (sem cursor readline, sem histórico)
- 'quiet', como modo plain, porém sem feedback
- 'secret', modo especial para leitura de senha
- `null` (padrão), com readline e histórico de entrada

Veja também

- [Biblioteca color](#)
- [Biblioteca curses](#)
- [Biblioteca elapsed](#)

curses

curses

Interface de usuário baseada em terminal (wrapper ncurses)

embora seja um wrapper, o FatScript tem sua própria maneira de abordar a interface do usuário no terminal, que pode diferir em alguns aspectos da biblioteca curses

Importação

```
_ <- fat.curses
```

Methods

Nome	Assinatura	Breve descrição
box	(p1: Scope, p2: Scope): Void	Desenhar quadrado de pos1 a pos2
clear	(): Void	Limpar buffer de tela
refresh	(): Void	Renderizar buffer de tela
getMax	(): Scope	Retorna tamanho da tela como x, y
printAt	(pos: Scope, msg: Any, width: Number = 0): Void	Imprimir msg em { x, y } pos
makePair	(fg: Number = 0, bg: Number = 0): Number	Criar um par de cores
usePair	(pair: Number): Void	Aplicar par de cores
frameTo	(cols: Number, rows: Number)	Alinhar área ao centro da tela
readKey	(): Text	Retorna tecla pressionada
readText	(pos: Scope, width: Number, prev: Text = 0): Text	Inicia caixa de texto
flushKeys	(): Void	Limpar buffer de entrada
endCurses	(): Void	Encerrar o modo curses

as posições (pos) estão no formato { x: Number, y: Number }

os métodos nesta biblioteca **não garantem** a segurança de threads em cenários assíncronos, ou utilize a thread principal **ou então** um único [worker](#) para renderizar atualizações no console

Notas de uso

Qualquer método desta biblioteca, exceto `getMax` e `endCurses`, iniciará o modo curses se ainda não tiver iniciado. Note que métodos como `log`, `stderr` e `input` da biblioteca [console](#) chamarão `endCurses` implicitamente. No entanto, `moveTo`, `print` e `flush` não irão alterar o modo de saída e podem ser combinados com métodos curses, o que pode ser útil em algumas circunstâncias.

As letras x e y representam coluna e linha, respectivamente, ao chamar `printAt`, onde { 0, 0 } é o canto superior esquerdo e o resultado de `getMax` é apenas a primeira coordenada fora do canto inferior direito.

caracteres especiais em curses só funcionam se um [locale](#) UTF-8 puder ser definido

makePair

Você pode importar a biblioteca [color](#) para usar nomes de cores e criar uma combinação de primeiro plano e plano de fundo (par). Passe `NULL` para aplicar a cor padrão no parâmetro desejado.

usePair

A entrada deste método deve ser um par de cores criado com o método `makePair`. Ele deixa este par habilitado até que você chame esta função novamente com um par diferente.

readKey

Este método não bloqueia e retorna `null` se `stdin` estiver vazio, caso contrário retornará um caractere por vez.

Chaves especiais podem ser detectadas e retornar palavras-chave como:

- teclas de seta:
 - up
 - down
 - left
 - right
- teclas de edição:
 - delete
 - backspace
 - enter
 - space
 - tab
 - backTab (shift+tab)
- teclas de controle:
 - pageUp
 - pageDown
 - home
 - end
 - insert
 - esc
- outras:
 - resize (janela do terminal foi redimensionada)

a detecção correta das teclas pode depender do contexto ou da plataforma

readText

Entra em modo captura de texto utilizando uma área demarcada por posição e largura da caixa de texto. Se o texto for maior que o espaço uma rolagem automática do texto é realizada. O texto completo é retornado ao pressionar `enter`, no entanto caso `ESC` seja pressionado é retornado `null`.

Veja também

- [Biblioteca color](#)
- [Biblioteca console](#)

failure

failure

Tratamento de erros e gerenciamento de exceções

Importação

```
_ <- fat.failure
```

Métodos

Nome	Assinatura	Breve descrição
trap	() : Void	Aplicar manipulador genérico de erro genérico
trapWith	(handler: Method) : Void	Definir um manipulador para erros no contexto
untrap	() : Void	Desarmar o manipulador para erros no contexto

Notas de uso

Quando um erro é criado se um manipulador de erro for encontrado, buscando do contexto de execução interno para externo, o manipulador que envolve a falha é invocado automaticamente com esse erro como argumento e o contexto de chamada é encerrado com o valor de retorno do manipulador de erro.

não é possível definir um manipulador para o escopo global

trapWith

Este método vincula um manipulador de erros ao contexto do site de chamada, por exemplo quando usado dentro de um método, ele apenas protegerá a lógica executada dentro do corpo desse método.

Exemplo

Defina um manipulador de erro que imprima o erro e saia:

```
console <- fat.console
system  <- fat.system
sdk     <- fat.sdk

simpleErrorHandler = (error) -> {
  console.log(error)
  sdk.printStackTrace(10)
  system.exit(system.failureCode)
}
```

Finalmente, use o método `trapWith` para atribuir o manipulador de erro:

```
failure <- fat.failure
failure.trapWith(simpleErrorHandler)
```

Trap it!

Você pode lidar com erros esperados ou deixar passar o inesperado:

```
failure <- fat.failure
_       <- fat.type.Error

MyError = Error

errorHandler = (error) -> {
  error == MyError => 0      # handle (expected)
  _                => error # pass through (unexpected)
}
```

failure

```
unsafeMethod = (n) -> {  
  failure.trapWith(errorHandler)  
  n < 10 ? MyError('arg is less than ten')  
  n - 10  
}
```

isso só funciona se a [opção](#) `-e` / `continuar` em caso de erro **não** estiver definida

Neste caso, o programa não travará se você chamar `unsafeMethod(5)`, mas se você comentar a linha `trapWith`, verá que ele trava com `MyError`.

Veja também

- [Error \(sintaxe\)](#)
- [Extensões do protótipo Error](#)

file

file

Operações de entrada e saída de arquivo

Importação

```
_ <- fat.file
```

Contribuições de Tipo

Nome	Assinatura	Breve descrição
FileInfo	(modTime: Epoch, size: Text)	Metadados do arquivo

Métodos

Nome	Assinatura	Breve descrição
basePath	() : Text	Extrair caminho onde o app foi chamado
exists	(path: Text): Boolean	Verificar se existe arquivo no caminho
read	(path: Text): Text	Ler arquivo do caminho (modo de texto)
write	(path: Text, src): Boolean	Escrever src no arquivo e retornar sucesso
append	(path: Text, src): Boolean	Acrescentar ao arquivo e retornar sucesso
remove	(path: Text): Boolean	Apagar o arquivo e retornar sucesso
isDir	(path: Text): Boolean	Verificar se o caminho é um diretório
mkDir	(path: Text, safe: Boolean)	Criar um diretório a directory
lsDir	(path: Text): List	Obter lista de arquivos em um diretório
stat	(path: Text): FileInfo	Obter metadados do arquivo

atualmente apenas o modo de texto é suportado (modo binário não é suportado), mas há uma [proposta](#) para adicionar suporte a isso no futuro

Notas de uso

read

Na exceção:

- registra o erro no `stderr`
- retorna `null`

`read` não pode ver "arquivos" embutidos, mas `readLib` da [biblioteca sdk](#) pode

write/append

Exceções:

- registra o erro no `stderr`
- retorna `false`

mkDir

Se `safe` estiver definido como `true`, o diretório receberá permissão 0700 em vez do padrão 0755, o que é menos protegido.

Veja também

- [Biblioteca csv](#)

file

- [Biblioteca.json](#)
-

http

http

Framework de manipulação HTTP

Importação

```
_ <- fat.http
```

Route

Uma rota é uma estrutura usada para mapear métodos HTTP para certos padrões de caminho, especificando qual código deve ser executado quando uma requisição é recebida. Cada rota pode definir um comportamento diferente para cada método HTTP (POST, GET, PUT, DELETE).

Construtor

Nome	Assinatura	Breve descrição
Route	(path: Text, post: Method, get: Method, put: Method, delete: Method)	Constrói um objeto Route

cada método implementado recebe um `HttpRequest` como argumento e deve retornar um objeto `HttpResponse`

HttpRequest

Um `HttpRequest` representa uma mensagem de requisição HTTP. Isso é o que seu servidor recebe de um cliente quando ele faz uma requisição ao seu servidor.

Construtor

Nome	Assinatura	Breve descrição
HttpRequest	(method: Text, path: Text, params: Scope, headers: List/Text, data: Text)	Constrói um objeto HttpRequest

HttpResponse

Um `HttpResponse` representa uma mensagem de resposta HTTP. Isso é o que um servidor envia de volta ao cliente em resposta a uma requisição HTTP.

Construtor

Nome	Assinatura	Breve descrição
HttpResponse	(status: Number, headers: List/Text, data: Text)	Constrói um objeto HttpResponse

Métodos

Nome	Assinatura	Breve descrição
setHeaders	(headers: List): Void	Definir cabeçalhos de solicitações
post	(url: Text, body, wait): HttpResponse	Criar/postar body para url
get	(url: Text, wait): HttpResponse	Ler/obter de url
put	(url: Text, body, wait): HttpResponse	Atualizar/colocar body na url
delete	(url: Text, wait): HttpResponse	Excluir em url
escape	(url: Text): Text	Codificar o texto para url-safe
unescape	(url: Text): Text	Decodificar texto url-safe
setName	(name: Text): Void	Definir agente/nome do servidor
listen	(port: Number, routes: List/Route)	Provedor de endpoint (modo servidor)

http

`body`: `Text` e `wait`: `Number` são sempre parâmetros opcionais, sendo que `wait` é o tempo máximo de espera e o padrão é 30.000ms (30 segundos)

Notas de uso

Modo cliente

Para dados em `HttpResponse`, você pode decodificar uma resposta JSON para o escopo `FatScript` usando o método `fromJson`, ou para `post` um escopo `FatScript` como JSON, você pode codificar usando o método `toJson`, ambos na biblioteca [fat.extra.json](#).

Os cabeçalhos padrão são:

```
[
  "Accept: application/json; charset=UTF-8"
  "Content-Type: application/json; charset=UTF-8"
]
```

Você pode definir cabeçalhos de solicitação personalizados da seguinte forma:

```
http <- fat.http
_ <- fat.extra.json

url = ...
token = ...
data = ...

http.setHeaders([
  "Accept: application/json; charset=UTF-8"
  "Content-Type: application/json; charset=UTF-8"
  "Authorization: Bearer " + token # cabeçalhos personalizado
])

http.post(url, toJson(data))
```

definir cabeçalhos substituirá completamente a lista anterior pela nova lista

Modo servidor

Lidando com respostas HTTP

O servidor `FatScript` lida automaticamente com os códigos de status HTTP comuns, como 200, 400, 404, 405, 500 e 501. Sendo 200 o padrão ao construir um objeto `HttpResponse`.

Além dos códigos de status retornados de forma automática, você também pode retornar explicitamente estes e outros códigos de status, como 201, 202, 203, 204, 205, 206, 301, 401 e 403, especificando o código de status no objeto `HttpResponse`, por exemplo: `HttpResponse(status = 401)`. Em todos os casos, quando aplicável, o servidor fornece corpos de resposta padrão em texto simples. No entanto, você tem a opção de substituir esses valores padrão e fornecer seus próprios corpos de resposta personalizados, quando necessário.

Ao lidar automaticamente com esses códigos de status e fornecer corpos de resposta padrão, o servidor `FatScript` simplifica o processo de desenvolvimento, ao mesmo tempo em que permite que você tenha controle sobre o conteúdo da resposta quando necessário.

não pertencendo a nenhum dos códigos anteriores, o servidor irá retornar o código 500

Veja um exemplo de um simples servidor HTTP de arquivos de texto:

```
_ <- fat.type.Text
file <- fat.file
http <- fat.http
{ Route, HttpRequest, HttpResponse } = http

# adapte para o local do conteúdo
basePath = '/home/user/contentFolder'

# restrito apenas a extensões de formato texto
getContentType = (path: Text): Text -> {
```

http

```
ext2 = path(-3..).toLowerCase
ext3 = path(-4..).toLowerCase
ext4 = path(-5..).toLowerCase

ext4 == '.html' => 'Content-Type: text/html'
ext3 == '.htm'  => 'Content-Type: text/html'
ext2 == '.js'   => 'Content-Type: application/javascript'
ext4 == '.json' => 'Content-Type: application/json'
ext3 == '.css'  => 'Content-Type: text/css'
ext2 == '.md'   => 'Content-Type: text/markdown'
ext3 == '.xml'  => 'Content-Type: application/xml'
ext3 == '.csv'  => 'Content-Type: text/csv'
ext3 == '.txt'  => 'Content-Type: text/plain'
ext4 == '.svg'  => 'Content-Type: image/svg+xml'
ext3 == '.rss'  => 'Content-Type: application/rss+xml'
ext4 == '.atom' => 'Content-Type: application/atom+xml'
-           => null
}

charset = '; charset=UTF-8'

routes: List/Route = [
  Route(
    '*'
    get = (request: HttpRequest): HttpResponse -> {
      path = basePath + request.path
      type = getContentType(path)

      !type           => HttpResponse(status = 403) # forbidden
      file.exists(path) => HttpResponse(data = file.read(path), headers = [type +
charset])
      -               => HttpResponse(status = 404) # not found
    }
  )
]

http.listen(8080, routes)
```

Observe que o FatScript atualmente não suporta manipulação de dados binários, mas há uma [proposta](#) para adicionar suporte a isso no futuro.

math

math

Operações e funções matemáticas

Importação

```
_ <- fat.math
```

Constantes

- e, logaritmo natural constante 2.71...
- maxInt, 9007199254740992
- minInt, -9007199254740992
- pi, razão do círculo para o seu diâmetro 3.14...

leia mais sobre [precisão numérica](#) no FatScript

Métodos

Nome	Assinatura	Breve descrição
abs	(x: Number): Number	Retorna o valor absoluto de x
ceil	(x: Number): Number	Retorna o menor inteiro $\geq x$
floor	(x: Number): Number	Retorna o maior inteiro $\leq x$
isInf	(x: Number): Boolean	Retorna true se x for infinito
isNan	(x: Any): Boolean	Retorna true se x não for um número
ln	(x: Number): Number	Retorna o logaritmo natural de x
random	(): Number	Retorna pseudo-aleatório, onde $0 \leq n < 1$
sqrt	(x: Number): Number	Retorna a raiz quadrada de x
sin	(x: Number): Number	Retorna o seno de x
cos	(x: Number): Number	Retorna o cosseno de x
asin	(x: Number): Number	Retorna o arco seno de x
acos	(x: Number): Number	Retorna o arco cosseno de x
atan	(x: Number, y = 1): Number	Retorna o arco tangente de x, y
max	(v: List/Number): Number	Retorna o valor máximo no vetor
min	(v: List/Number): Number	Retorna o valor mínimo no vetor
sum	(v: List/Number): Number	Retorna a soma do vetor

Exemplo

```
math <- fat.math # importação nomeada
math.abs(-52)   # retorna 52
```

Veja também

- [Number \(sintaxe\)](#)
- [Extensões do protótipo Number](#)
- [Biblioteca matemática estendida](#)

sdk

sdk

Utilitários do kit de desenvolvimento de software da Fry

esta é uma biblioteca especial que expõe alguns dos elementos internos do interpretador fry para serem usados como ferramentas de depuração (`ast`, `printStack`, `readLib`) ou blocos de construção para recursos avançados (`eval`, `getVersion`, `typeof`)

Importação

```
_ <- fat.sdk
```

Métodos

Nome	Assinatura	Breve descrição
<code>ast</code>	<code>(_): Void</code>	Imprimir árvore de sintaxe abstrata do nó
<code>stringify</code>	<code>(_): Text</code>	Converte nó em texto json
<code>eval</code>	<code>(_): Any</code>	Interpreta texto como programa FatScript
<code>getVersion</code>	<code>(): Text</code>	Retorno versão do fry
<code>printStack</code>	<code>(depth: Number): Void</code>	Imprimir pilha do contexto de execução
<code>readLib</code>	<code>(ref: Text): Text</code>	Retorna o código-fonte da biblioteca fry
<code>typeof</code>	<code>(_): Text</code>	Retorna o tipo do nó
<code>isMain</code>	<code>(): Boolean</code>	Está executando como principal ou módulo
<code>getMeta</code>	<code>(): Scope</code>	Retorna os metadados do interpretador

Exemplo

```
_ <- fat.sdk  
_ <- fat.console
```

```
print(readLib('fat.extra.csv')) # imprime a implementação da biblioteca csv
```

`readLib` não pode ver arquivos externos, mas `read` da [biblioteca file](#) pode

system

system

Operações e informações em nível de sistema

Importação

```
_ <- fat.system
```

Aliases

Nome	Tipo original	Breve descrição
ExitCode	Number	Status de saída ou código de retorno
CommandResult	Scope	Contém code (código) e out (saída)

Constantes

- successCode, 0: ExitCode
- failureCode, 1: ExitCode

Métodos

Nome	Assinatura	Breve descrição
args	() : List/Text	Retorna lista de argumentos passados pelo shell
exit	(code: Number): *	Sair do programa com o código de saída fornecido
getEnv	(var: Text): Text	Obter o valor da variável env por nome
shell	(cmd: Text): ExitCode	Executa cmd no shell, retorna o código de saída
capture	(cmd: Text): CommandResult	Captura a saída da execução de cmd
fork	(args: List/Text, out: Text = \emptyset)	Inicia um processo em segundo plano, retorna PID
kill	(pid: Number): Void	Envia SIGTERM para o processo pelo PID
getLocale	() : Text	Obter configuração de localidade atual
setLocale	(cmd: Text): Number	Definir configuração de localidade atual
getMacId	() : Text	Obter identificador da máquina (endereço MAC)
setKey	(key: Text): Void	Definir chave para pacotes ofuscados
setMem	(mem: Number): Void	Definir limite de memória (contagem de nós)
runGC	() : Number	Rodar o GC, retorna transcorrido em milissegundos

Notas de uso

Atenção!

É importante agir com cautela e responsabilidade ao utilizar os métodos `getEnv`, `shell`, `capture`, `fork` e `kill`. A biblioteca `system` oferece a capacidade de executar comandos diretamente do sistema operacional, o que pode introduzir riscos de segurança se não forem utilizadas com cuidado.

Para mitigar vulnerabilidades, evite utilizar a entrada do usuário diretamente na construção de comandos passados para esses métodos. A entrada do usuário deve ser validada para prevenir ataques de injeção de comandos e outras violações de segurança.

setKey

Use preferencialmente no arquivo `.fryrc` assim:

```
_ <- fat.system
setKey('secret') # irá codificar e decodificar pacotes com esta chave
```

system

Veja mais sobre [ofuscação](#).

setMem

Use preferencialmente no arquivo `.fry` assim:

```
_ <- fat.system  
setMem(5000) # ~2mb
```

Veja mais sobre [gerenciamento de memória](#).

get/set locale

O interpretador `fry` tentará inicializar o locale `LC_ALL` para `C.UTF-8` e, se esse locale não estiver disponível no sistema, tentará usar `en_US.UTF-8`, caso contrário, usará o locale padrão.

Veja mais sobre [nomes de localidade](#).

a configuração de localidade aplica-se apenas ao aplicativo e não é mantida após a saída do `fry`

time

time

Manipulação de hora e data

Importação

```
_ <- fat.time
```

Aliases

Nome	Tipo original	Breve descrição
Epoch	Number	Tempo de época do Unix em milissegundos

[tipo Number](#) é importado automaticamente com esta importação

Métodos

Nome	Assinatura	Breve descrição
setZone	(offset: Number): Void	Definir fuso em milissegundos
getTimeZone	() : Number	Obter diferença de fuso atual
now	() : Epoch	Obtenha o UTC atual em Epoch
format	(date: Text, fmt: Text = \emptyset): Epoch	Converter data para Epoch
parse	(date: Text, fmt: Text = \emptyset): Epoch	Converter Epoch em formato de data
wait	(ms: Number): Void	Aguardar milissegundos (suspend)

Notas de uso

Epoch

No FatScript, o tempo é representado como um tipo aritmético para que você possa fazer contas.

Você pode obter o tempo decorrido entre tempo1 e tempo2 como:

```
decorrido = tempo2 - tempo1
```

Você também pode verificar se tempo2 acontece após tempo1, simplesmente assim:

```
tempo2 > tempo1
```

format

Formata a data em texto como "%Y-%m-%d %H:%M:%S.milliseconds" (padrão), quando `fmt` é omitido.

milissegundos só podem ser transformados no formato padrão, caso contrário, a precisão é de até segundos

argumento fmt

A especificação de formato é um texto contendo uma sequência de caracteres especiais chamadas especificações de conversão, cada uma das quais é introduzida por um caractere '%' e terminada por algum outro caractere conhecido como especificador de conversão. Todos os outros caracteres são tratados como texto comum.

Especificador	Significado
%a	Nome abreviado do dia da semana
%A	Nome completo do dia da semana
%b	Nome do mês abreviado
%B	Nome completo do mês

Especificador	Significado
%c	Data/Hora no formato da localidade
%C	Número do século [00-99], o ano dividido por 100 e truncado para um número inteiro
%d	Dia do mês [01-31]
%D	Formato de data, igual a %m/%d/%y
%e	O mesmo que %d, exceto que um único dígito é precedido por um espaço [1-31]
%g	Parte do ano de 2 dígitos da data da semana ISO [00,99]
%F	Formato de data ISO, igual a %Y-%m-%d
%G	Parte do ano de 4 dígitos da data da semana ISO
%h	O mesmo que %b
%H	Hora no formato de 24 horas [00-23]
%I	Hora em formato de 12 horas [01-12]
%j	Dia do ano [001-366]
%m	Mês [01-12]
%M	Minuto [00-59]
%n	Caractere de nova linha
%p	Cadeia AM ou PM
%r	Hora no formato AM-PM da localidade
%R	Formato de 24 horas sem segundos, igual a %H:%M
%S	Segundo [00-61], o intervalo de segundos permite um segundo bissexto e um segundo bissexto duplo
%t	Caractere de tabulação
%T	Formato de 24 horas com segundos, igual a %H:%M:%S
%u	Dia da semana [1,7], segunda-feira é 1 e domingo é 7
%U	Número da semana do ano [00-53], domingo é o primeiro dia da semana
%V	Número da semana ISO do ano [01-53]. Segunda-feira é o primeiro dia da semana. Se a semana contendo 1º de janeiro tiver quatro ou mais dias no ano novo, será considerada a semana 1. Caso contrário, será a última semana do ano anterior e o ano seguinte será a semana 1 do ano novo.
%w	Dia da semana [0,6], domingo é 0
%W	Número da semana do ano [00-53], segunda-feira é o primeiro dia da semana
%x	Data no formato da localidade
%X	Hora no formato da localidade
%y	Ano de 2 dígitos [00,99]
%Y	Ano de 4 dígitos (pode ser negativo)
%z	String de deslocamento UTC com formato +HHMM ou -HHMM
%Z	Nome do fuso horário
%%	Caractere %

Sob o capô `format` usa C's [strftime](#) e `parse` usa C's [strptime](#), mas a tabela de especificação de formato acima se aplica praticamente nos dois sentidos.

zCode

zCode

Métodos de codificação de dados, hash e uuid

Importação

```
_ <- fat.zcode
```

Métodos

Nome	Assinatura	Breve descrição
getHash	(msg: Text): Number	Obter hash de texto de 32 bits
getUuid	(): Text	Gerar um UUID (versão 4)
encrypt	(msg: Text, key: Text = "): Text	Codificar msg para zCode usando key
decrypt	(msg: Text, key: Text = "): Text	Decodificar msg zCoded usando key

Notas de uso

Você pode omitir ou passar uma chave (key) em branco ' ' para usar a chave padrão.

Atenção!

Embora zCode torne o texto codificado "não legível por humanos", esse esquema não é criptograficamente seguro! NÃO o utilize sozinho para proteger dados!

Se pareado com uma chave personalizada que não esteja armazenada junto com a mensagem, pode oferecer alguma proteção de dados.

Conformidade do método UUID

Um UUID, ou Universally Unique Identifier, é um número de 128 bits usado para identificar objetos ou entidades em sistemas de computador. A implementação fornecida gera UUIDs aleatórios como texto que segue o formato da versão 4 da especificação RFC 4122, mas não adere estritamente à aleatoriedade criptograficamente segura necessária. Na prática, o risco de colisão tem uma probabilidade extremamente baixa e é muito improvável de ocorrer, e para a maioria das aplicações pode ser considerado bom o bastante.

Veja também

- [Biblioteca hex](#)

type._

pacote type

Extensões do protótipo para [tipos nativos](#):

- [Void](#)
- [Boolean](#)
- [Number](#)
- [Text](#)
- [Method](#)
- [List](#)
- [Scope](#)
- [Error](#)

FatScript **não** carrega essas definições automaticamente no escopo global, portanto você deve **explicitamente** [importar](#) quando necessário

Importando

Se você quiser disponibilizar todos eles de uma só vez, basta escrever:

```
_ <- fat.type._
```

...ou importe um por um, conforme necessário, por exemplo:

```
_ <- fat.type.List
```

características comum

Todos os tipos neste pacote suportam os seguintes métodos de protótipo:

- `apply` (construtor)
- `isEmpty`
- `nonEmpty`
- `size`
- `toText`

Veja também

- [Tipos \(sintaxe\)](#)

Void

Void

Extensões do protótipo Void

Importação

```
_ <- fat.type.Void
```

Construtor

Nome	Assinatura	Breve descrição
Void	(val: Any)	Retorna nulo, apenas ignora o argumento

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	()	Boolean Retorna verdadeiro, sempre
nonEmpty	()	Boolean Retorno falso, sempre
size	()	Number Retorna 0, sempre
toText	()	Text Retorna 'null' como texto

Exemplo

```
_ <- fat.type.Void  
x.isEmpty # true, já que x não foi declarado
```

Veja também

- [Void \(sintaxe\)](#)
- [Pacote type](#)

Boolean

Boolean

Extensões do protótipo Boolean

Importação

```
_ <- fat.type.Boolean
```

Construtor

Nome	Assinatura	Breve descrição
Boolean	(val: Any)	Coage o valor para booleano

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	()	Boolean Retorna verdadeiro se falso
nonEmpty	()	Boolean Retorna falso se verdadeiro
size	()	Number Retorna 1 se verdadeiro, 0 se falso
toText	()	Text Retorna 'true' ou 'false' como texto

Exemplos

```
_ <- fat.type.Boolean
```

```
x = true
x.isEmpty # falso, já que x é verdadeiro
```

```
Boolean('false') # retorna true, porque o texto é não-vazio
Boolean('')      # retorna falso, porque é vazio
```

note que o construtor não tenta converter o valor do texto, o que é consistente com as avaliações de controle de fluxo, e você pode usar um simples [case](#) se precisar fazer conversão de texto para booleano

Veja também

- [Boolean \(sintaxe\)](#)
- [Pacote type](#)

Number

Number

Extensões do protótipo Number

Importação

```
_ <- fat.type.Number
```

Construtor

Nome	Assinatura	Breve descrição
Number	(val: Any)	Texto para número ou tamanho da coleção

realiza a conversão de texto para número assumindo a base decimal

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	() : Boolean	Retorna verdadeiro se zero
nonEmpty	() : Boolean	Retorna verdadeiro se não-zero
size	() : Number	Retorna valor absoluto, igual a math.abs
toText	() : Text	Retorna número como texto
format	(fmt: Text): Text	Retorna número como texto formatado

Exemplo

```
_ <- fat.type.Number
x = Number('52') # número: 52
x.toText         # texto: '52'
```

format

O método `format` é usado para converter números em strings de várias maneiras. A estrutura básica de um especificador de formato é `%[flags][width][.precision][type]`. Aqui está o que cada um destes componentes significa:

- `flags` são caracteres opcionais que controlam o comportamento específico de formatação. Por exemplo, `0` pode ser usado para preenchimento com zeros e `-` para justificação à esquerda.
- `width` é um número inteiro que especifica o número mínimo de caracteres a serem impressos. Se o valor a ser impresso for mais curto do que este número, o resultado é preenchido com espaços em branco ou zeros, dependendo da flag utilizada.
- `precision` é um número opcional que segue um `.` que especifica o número de dígitos a serem impressos após o ponto decimal.
- `type` é um caractere que especifica como o número deve ser representado. Os tipos comuns são `f` (notação de ponto fixo), `e` (notação exponencial), `g` (fixo ou exponencial dependendo da magnitude do número) e `a` (notação de ponto flutuante hexadecimal).

Exemplos:

- `%5.f`: Isso imprimirá o número com uma largura total de 5 caracteres, sem dígitos após o ponto decimal (porque a precisão é `f`, que significa ponto fixo, mas nenhum número segue o ponto). Será justificado à direita porque nenhuma flag `-` é usada.
- `%05.f`: Semelhante ao anterior, mas como a flag `0` é usada, os espaços vazios serão preenchidos com zeros.
- `%8.2f`: Isso imprimirá o número com uma largura total de 8 caracteres, com 2 dígitos após o ponto decimal.

Number

- %-8.2f: Semelhante ao anterior, mas o número será justificado à esquerda por causa da flag -.
- %.2e: Isso imprimirá o número usando notação exponencial, com 2 dígitos após o ponto decimal.
- %.2a: Isso imprimirá o número usando notação de ponto flutuante hexadecimal, com 2 dígitos após o ponto hexadecimal.
- %.2g: Isso imprimirá o número em notação de ponto fixo ou exponencial, dependendo da sua magnitude, com no máximo 2 dígitos significativos.

Veja também

- [Number \(sintaxe\)](#)
- [Biblioteca math](#)
- [Pacote type](#)

Text

Text

Extensões do protótipo Text

Importação

```
_ <- fat.type.Text
```

Construtor

Nome	Assinatura	Breve descrição
Text	(val: Any)	Coage valor para texto, igual a .toText

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	() : Boolean	Retorna verdadeiro se o comprimento for zero
nonEmpty	() : Boolean	Retorna verdadeiro se comprimento não-zero
size	() : Number	Comprimento do texto de retorno
toText	() : Text	Forçar interpolação de texto
replace	(old: Text, new: Text): Text	Substitua old por new
indexOf	(frag: Text): Number	Obter índice de fragmento, -1 se ausente
contains	(frag: Text): Boolean	Verifique se o texto contém fragmento
split	(sep: Text): List/Text	Dividir texto por set em lista
toLowerCase	() : Text	Retorna a versão minúscula do texto
toUpperCase	() : Text	Retorna a versão maiúscula do texto
trim	() : Text	Retorna a versão aparada do texto
match	(regex: Text): Boolean	Retorna se o texto corresponde à regex

Exemplo

```
_ <- fat.type.Text
x = 'banana'
x.size                # retorna 6
x.replace('ana', 'nquete'); # produz 'banquete'
```

Veja também

- [Text \(sintaxe\)](#)
- [Pacote type](#)
- [Biblioteca regex](#)

Method

Method

Extensões do protótipo Method

Importação

```
_ <- fat.type.Method
```

Construtor

Nome	Assinatura	Breve descrição
Method	(val: Any)	Envolve val em um método

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	()	Boolean Retorna falso, sempre
nonEmpty	()	Boolean Retorna verdadeiro, sempre
size	()	Number Retorna 1, sempre
toText	()	Text Retorna o literal de texto 'Method'
arity	()	Number Retorna a aridade do método

Exemplo

```
_ <- fat.type.Method  
x = (): Number -> 3  
x.toText # retorna 'Method'
```

Veja também

- [Method \(sintaxe\)](#)
- [Pacote type](#)

List

List

Extensões do protótipo List

Importação

```
_ <- fat.type.List
```

Construtor

Nome	Assinatura	Breve descrição
List	(val: Any)	Envolver val em uma lista

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	() : Boolean	Retorna verdadeiro se o comprimento for zero
nonEmpty	() : Boolean	Retorna verdadeiro para comprimento não zero
size	() : Number	Retorna o comprimento da lista
toText	() : Text	Retorna o literal de texto 'List'
join	(sep: Text): Text	Junta lista com separador em texto
flatten	() : List	Achata lista de listas em uma plana
find	(p: Method): Any	Retorna a primeira correspondência ou nulo
contains	(p: Method): Boolean	Checa se algum item corresponde ao predicado
filter	(p: Method): List	Retorna sub-lista que corresponde ao predicado
reverse	() : List	Retorna uma cópia invertida da lista
shuffle	() : List	Retorna uma cópia embaralhada da lista
unique	() : List	Retorna itens únicos da lista
sort	() : List	Retorna uma cópia ordenada da lista
sortBy	(key: Any): List	Retorna uma cópia ordenada da lista *
indexOf	(item: Any): Number	Retorna índice do item, -1 se ausente

Exemplo

```
_ <- fat.type.List
x = [ 'a', 'b', 'c' ]
x.size # retorna 3
```

Ordenação

Os métodos `sort` e `sortBy` utilizam o algoritmo quicksort com seleção aleatória de pivô. Essa abordagem é conhecida por sua eficiência e geralmente opera com uma complexidade de tempo média de $O(n \log n)$, tornando-a adequada para a maioria dos conjuntos de dados. Embora a ordenação seja geralmente eficiente, a ordem original de elementos equivalentes pode não ser garantida, o que pode ser um problema quando a sequência inicial de elementos semelhantes é importante (conhecido como ordenação estável). Para casos de uso mais avançados, considere a possibilidade de dividir seu conjunto de dados em grupos primeiro.

`sortBy` aceita um parâmetro textual para `key` se for uma lista de `Scope` ou um parâmetro numérico caso seja uma lista de `List` (matriz), representando o índice

Veja também

- [List \(sintaxe\)](#)
 - [Pacote type](#)
-

Scope

Scope

Extensões do protótipo Scope

Importação

```
_ <- fat.type.Scope
```

Construtor

Nome	Assinatura	Breve descrição
Scope	(val: Any)	Envolver val em um escopo

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	()	Boolean Retorna verdadeiro se o tamanho for zero
nonEmpty	()	Boolean Retorna verdadeiro para tamanho não zero
size	()	Number Retorna o número de entradas do escopo
toText	()	Text Retorna o literal de texto 'Scope'

Exemplo

```
_ <- fat.type.Scope  
x = { num = 12, prop = 'outra' }  
x.size # retorna 2
```

Veja também

- [Scope \(sintaxe\)](#)
- [Pacote type](#)

Error

Error

Extensões do protótipo Error

Importação

```
_ <- fat.type.Error
```

Aliases

- `KeyError`
- `IndexError`
- `CallError`
- `TypeError`
- `AssignError`
- `ValueError`

Construtor

Nome	Assinatura	Breve descrição
Error	(val: Any)	Retornar val coagido para texto como erro

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	()	Boolean Retorna verdadeiro, sempre
nonEmpty	()	Boolean Retorna falso, sempre
size	()	Number Retorna 0, sempre
toText	()	Text Retorna o texto do erro

Exemplo

```
_ <- fat.type.Error
x = Erro('ops')
x.toText # retorna "Erro: ops"

# ...ou algo inesperado
e = undeclared.item # gera erro
e.toText           # retorna "can't resolve scope of 'item'"
```

Veja também

- [Biblioteca failure](#)
- [Error \(sintaxe\)](#)
- [Pacote type](#)

extra._

pacote extra

Utilitários implementados em FatScript puro:

- [csv](#) - Codificador e decodificador rudimentar de CSV
- [Date](#) - Gerenciamento de calendário e datas
- [Duration](#) - Construtor de duração em milissegundos
- [elapsed](#) - Calculadora de tempo decorrido
- [HashMap](#) - Armazenamento rápido de chave-valor
- [hex](#) - Codificador e decodificador hexadecimal
- [json](#) - Codificação e armazenamento de dados JSON
- [Logger](#) - Suporte ao registro de logs
- [mathex](#) - Biblioteca matemática estendida
- [Memo](#) - Classe de utilidade de memoização genérica
- [regex](#) - Padrões comuns de expressões regulares
- [Sound](#) - Interface de reprodução de som
- [util](#) - Outras utilidades aleatórias
- [xml](#) - Analisador e gerador de XML simplificado

Importando

Se você quiser disponibilizar todos eles de uma só vez, basta escrever:

```
_ <- fat.extra._
```

...ou importe um por um, conforme necessário, por exemplo:

```
_ <- fat.extra.json
```

Nota do desenvolvedor

Atualmente, a maioria desses utilitários não são otimizados para recursos ou desempenho.

A intenção aqui era mais fornecer recursos simples, como modelos básicos que podem ser extraídos via [readLib](#), para que qualquer desenvolvedor com requisitos específicos tenha um ponto de partida para suas próprias implementações.

CSV

CSV

Codificador e decodificador CSV rudimentar

Importação

```
_ <- fat.extra.csv
```

[pacote de tipos](#) é automaticamente importado com esta importação

Variáveis

Essas entradas são definidas via importação e podem ser atualizadas posteriormente para configurar o comportamento dos métodos da biblioteca:

- `csvSeparator`, o padrão é vírgula `,`
- `csvReplacement`, o padrão é texto vazio `' '`

Métodos

Nome	Assinatura	Breve descrição
<code>toCsv</code>	(header: List/Text, rows: List/Scope): Text	Criar csv a partir de linhas
<code>fromCsv</code>	(csv: Text): List/Scope	Converter csv para lista de linhas

Uso

`toCsv`

Código de exemplo:

```
_ <- fat.extra.csv
```

```
headers = [ 'name', 'stock', 'sale' ]
```

```
data = [
  { sale = true,  stock = 52, name = 'Apple' }
  { sale = false, stock = 35, name = 'Orange' }
  { sale = true,  stock = 24, name = 'Banana' }
]
```

```
toCsv(headers, data) # name,stock,sale\nApple,52,true\nOrange,35,false...
```

`csvReplacement` é usado por `toCsv` como substituto no caso de um `csvSeparator` ser encontrado dentro de um texto sendo codificado

`fromCsv`

Código de exemplo:

```
csvData = fromCsv(data) # Lista/Escopto de dados originais
```

escapar entrada contendo `csvSeparator` com aspas não é suportado

Veja também

- [Pacote extra](#)

Date

Date

Gerenciamento de calendário e datas

operações como adição e subtração de dias, meses e anos, garantindo o tratamento preciso de várias complexidades relacionadas a datas, como anos bissextos e cálculos de final de mês

Importação

```
_ <- fat.extra.Date
```

[biblioteca time](#), [biblioteca math](#), [tipo Error](#), [tipo Text](#), [tipo List](#), [tipo Number](#), [tipo Duration](#) são automaticamente importados com esta importação

Tipo Date

Date oferece uma solução abrangente para o gerenciamento de datas, incluindo anos bissextos e horário do dia.

Propriedades

- `year`: Número - Ano da data
- `month`: Número - Mês da data
- `day`: Número - Dia da data
- `tms`: Milissegundos - Horário do dia em milissegundos

valor padrão aponta para: 1 de janeiro de 1970

Membros do Protótipo

Nome	Assinatura	Breve descrição
<code>fromEpoch</code>	(ems: Epoch): Date	Cria uma instância a partir de um epoch
<code>isLeapYear</code>	(year: Número): Boolean	Determina se um ano é bissexto
<code>normalizeMonth</code>	(month: Número): Número	Normaliza o número do mês
<code>daysInMonth</code>	(year: Número, month: Número): Número	Retorna o número de dias no mês de um ano
<code>isValid</code>	(year, month, day, tms): Boolean	Valida os componentes da data
<code>truncate</code>	(): Date	Trunca o horário do dia
<code>toEpoch</code>	(): Epoch	Converte a instância para tempo em epoch
<code>addYears</code>	(yearsToAdd: Número): Date	Adiciona anos à data
<code>addMonths</code>	(monthsToAdd: Número): Date	Adiciona meses à data
<code>addWeeks</code>	(weeksToAdd: Número): Date	Adiciona semanas à data
<code>addDays</code>	(daysToAdd: Número): Date	Adiciona dias à data

Exemplos de uso

```
_ <- fat.extra.Date

# Criar uma instância de Data
myDate = Date(2023, 1, 1)

# Adicionar um ano à data
newDate = myDate.addYears(1)

# Adicionar duas semanas à uma data
datePlusTwoWeeks = myDate.addWeeks(2)

# Criar uma Data a partir de um tempo em epoch (em milissegundos)
# o resultado é influenciado pelo fuso horário atual, veja: time.setZone
```

Date

```
epochTime = 1672531200000  
dateFromEpoch = Date.fromEpoch(Epoch(epochTime))
```

```
# Converter uma data para tempo em epoch  
epochFromDate = myDate.toEpoch
```

Duration

Duration

Construtor de duração em milissegundos

No FatScript, o tempo é nativamente expresso em milissegundos, e esse tipo fornece uma maneira simples de expressar diferentes magnitudes de tempo em Millis.

Importação

```
_ <- fat.extra.Duration
```

Aliases

Nome	Tipo Original	Breve descrição
Millis	Number	Tempo em milissegundos

Construtor

Nome	Assinatura	Breve descrição
Duration	(val: Number)	Cria um conversor de duração em Millis

Membros do protótipo

Nome	Assinatura	Breve descrição
nanos	() : Millis	Interpretar valor como nanossegundos
micros	() : Millis	Interpretar valor como microssegundos
millis	() : Millis	Interpretar valor como milissegundos
seconds	() : Millis	Interpretar valor como segundos
minutes	() : Millis	Interpretar valor como minutos
days	() : Millis	Interpretar valor como dias
weeks	() : Millis	Interpretar valor como semanas
months	() : Millis	Interpretar valor como meses (aprox.)
years	() : Millis	Interpretar valor como anos (aprox.)

Exemplo

```
_ <- fat.extra.Duration
time <- fat.time

cincoSegundos = Duration(5).segundos
time.wait(cincoSegundos) # pausa a execução do thread por 5 segundos
```

elapsed

elapsed

Calculadora de tempo decorrido

Importação

```
_ <- fat.extra.elapsed
```

Métodos

Nome	Assinatura	Breve descrição
getElapsed	(since: Epoch): Text	Retorna o tempo decorrido como texto
showElapsed	(label: Text, since: Epoch): Text	Linha de log com tempo decorrido

[biblioteca time](#) é importada automaticamente com esta importação

Notas de uso

Exemplo:

```
_ <- fat.extra.elapsed
start = now() # veja: biblioteca de tempo
ms = 300
wait(ms)
showElapsed('levou', start) # imprime 'levou 300 ms'
```

elapsed arredondará automaticamente milissegundos para segundos, minutos ou horas, mantendo apenas a parte inteira

Veja também

- [Biblioteca console](#)
- [Pacote extra](#)

HashMap

HashMap

Armazenamento otimizado em memória de par chave-valor, servindo como um substituto de melhor desempenho para a implementação padrão do Scope, projetado para lidar eficientemente com grandes conjuntos de dados.

os ganhos de velocidade vem em detrimento de um maior uso de memória

Importação

```
_ <- fat.extra.HashMap
```

Construtor

Nome	Assinatura	Breve descrição
HashMap	(capacity: Number = 97)	Cria um HashMap com uma capacidade especificada

Membros do protótipo

Nome	Assinatura	Breve descrição
set	(key: Text, value: Any): Any	Define um par chave-valor no HashMap
get	(key: Text): Any	Obtém o valor associado a uma chave
keys	(): List/Text	Retorna uma lista de todas as chaves do HashMap
values	(): List/Any	Retorna uma lista de todos os valores do HashMap

Exemplo

```
_ <- fat.extra.HashMap

hmap = HashMap()
hmap.set('key1', 'value1')

hmap.get('key1') # retorna 'value1'
hmap.keys       # retorna [ 'key1' ]
hmap.values     # retorna [ 'value1' ]
```

hex

hex

Codificador e decodificador hexadecimal

Importação

```
_ <- fat.extra.hex
```

Constantes

- `hexDigits`, '0123456789abcdef'

Métodos

Nome	Assinatura	Breve descrição
<code>toHex</code>	(dec: Number): Text	Codificar número decimal para hexadecimal
<code>fromHex</code>	(hex: Text): Number	Decodificar texto hexadecimal para número

o valor máximo a ser codificado / decodificado está limitado a [precisão numérica](#)

Notas de uso

Exemplo:

```
toHex(128)           # Text: '80'  
fromHex('FFFFFF')  # Number: 16777215
```

`toHex` não implementa preenchimento de zero à esquerda, mas isso pode ser feito com o uso do método `padLeft` na biblioteca [util](#)

Veja também

- [Pacote extra](#)

json

json

Codificação e armazenamento de dados JSON

Importação

```
_ <- fat.extra.json
```

[biblioteca file](#), [biblioteca sdk](#), [biblioteca zcode](#), [tipo Error](#), [tipo Text](#), [tipo Void](#) e [tipo Method](#) são automaticamente importados com esta importação

Aliases

Nome	Tipo original	Breve descrição
FileError	Error	Erro personalizado para operações com arquivos

Mixins

A biblioteca `json` introduz dois tipos de mixin: `Storable` e `EncryptedStorable`

Storable

O mixin `Storable` fornece métodos para armazenar e recuperar objetos no sistema de arquivos usando serialização JSON.

Membros do Protótipo

Nome	Assinatura	Breve descrição
list	() : List<Text>	Obtém lista de ids para instâncias armazenadas
load	(id: Text) : Any	Carrega um objeto do sistema de arquivos
save	() : Boolean	Salva a instância do objeto atual
erase	() : Boolean	Exclui o arquivo associado ao id

os métodos `load` e `save` emitem `FileError` em caso de falha

EncryptedStorable

Estende `Storable` com capacidades de criptografia para um armazenamento de dados mais seguro. Requer uma implementação do método `getEncryptionKey`.

Métodos avulsos

Nome	Assinatura	Breve descrição
toJson	(_): Text	Cria um json a partir de tipos nativos
fromJson	(_): Any	Converte um json para tipos nativos

Notas de uso

"Com grandes poderes vêm grandes responsabilidades" -Peter Parker

Como o FatScript aceita alternativamente [sintaxe semelhante a JSON](#), `fromJSON` na verdade usa o analisador/lexer interno do FatScript via [eval](#), que é extremamente rápido, mas pode ou não produzir exatamente o que se espera de um conversor JSON.

Por exemplo, uma vez que o fragmento abaixo é analisado, já que `null` em FatScript é ausência de valor, não haveria declarações de entrada para "prop":

```
"prop": null
```

Portanto, ler com `fromJson` e escrever de volta com `toJson` não é necessariamente uma operação idempotente.

Atenção!

O método `fromJson` deve ser perfeitamente correto e seguro para leitura de arquivos de configuração ou leitura de dados armazenados via `toJson`.

No entanto, como `fromJson` ingere dados via `sdk.eval`, um arquivo especialmente criado pode implementar um programa `FatScript` e executar código arbitrário!

Se estiver lendo arquivos JSON de uma fonte desconhecida, por segurança, você deve criar seu próprio conversor orientado à segurança.

se você escreveu um conversor JSON alternativo em `FatScript` e gostaria de compartilhar a referência aqui, consulte o documento [contributing](#)

Mixins

Instâncias de `Storable` estão vinculadas a arquivos por um `id` único. Se não fornecido, a implementação padrão gerará um [UUID aleatório](#).

Exemplo

```
_ <- fat.extra.json

# Defina um tipo que inclua Storable (ou EncryptedStorable)
User = (
  Storable # Inclui o mixin Storable

  # EncryptedStorable # implementação alternativa
  # getEncryptionKey = (): Text -> '3ncryp1ptM3' # poderia obter via KMS ou
  configuração

  ## Argumentos
  name: Text
  email: Text

  # Os setters retornam uma nova cópia imutável da instância com o campo atualizado
  setName = (name: Text) -> $self + User * { name }
  setEmail = (email: Text) -> $self + User * { email }
)

# Cria uma nova instância de usuário
newUser = User('Jane Doe', 'jane.doe@example.com')

# Salva o novo usuário
newUser.save

# Atualiza as informações do usuário e salva as alterações
updatedUser = newUser
  .setName('Jane Smith')
  .setEmail('jane.smith@example.com')
updatedUser.save

# Lista todos os usuários salvos
userIds = User.list

# Carrega um usuário do sistema de arquivos
userId = userIds(0) # ...ou newUser.id
loadedUser = User.load(userId)

# Exclui os dados do usuário do sistema de arquivos
loadedUser.erase # ...ou User.erase(userId)
```

Veja também

- [Pacote extra](#)

Logger

Logger

Suporte ao registro de logs

desde logs simples em console a registro baseado em arquivos

Importação

```
_ <- fat.extra.Logger
```

[biblioteca console](#), [biblioteca color](#), [biblioteca file](#), [biblioteca time](#), [biblioteca sdk](#) e [biblioteca de tipos](#) são automaticamente importadas com esta importação

Tipo Logger

Logger oferece capacidades de registro de logs personalizáveis com vários níveis e formatos.

Propriedades

- `level`: Text (padrão 'debug') - Nível de log
- `showTime`: Boolean (padrão verdadeiro) - Indicador para exibir carimbos de hora

níveis válidos: 'debug', 'info', 'warn', 'error'

Membros do Protótipo

Nome	Assinatura	Breve descrição
<code>setLevel</code>	(level: Text)	Define o nível de log
<code>setShowTime</code>	(showTime: Boolean)	Alterna a exibição de carimbos de hora nos logs
<code>asMessage</code>	(level: Text, args: Scope): Texto	Formata mensagens de log (pode ser substituído)
<code>log</code>	(msg: Any, fg: Number)	Registra mensagens (pode ser substituído)

Funções de Log

- `debug(_1, _2, _3, _4, _5)`: Registra uma mensagem de debug
- `info(_1, _2, _3, _4, _5)`: Registra uma mensagem informativa
- `warn(_1, _2, _3, _4, _5)`: Registra uma mensagem de aviso
- `error(_1, _2, _3, _4, _5)`: Registra uma mensagem de erro

Subtipos

BoringLogger

- Herda de `Logger`
- Substitui `log` para emitir texto simples sem cor

FileLogger

- Herda de `Logger`
- Propriedades Adicionais:
 - `logfile`: Texto (padrão 'log.txt') - arquivo para registro de logs
- Substitui `log` para anexar mensagens a um arquivo

Exemplo de uso

```
_ <- fat.extra.Logger
```

```
# Crie uma instância com configurações personalizadas
```

Logger

```
myLogger = Logger(level = 'info', showTime = false)

# Registra uma mensagem informativa
myLogger.info('Esta é uma mensagem informativa.')

# Crie um FileLogger para registrar mensagens em um arquivo
fileLogger = FileLogger('meuLog.txt')
fileLogger.info('Registrado no arquivo.')
```

mathex

mathex

Biblioteca matemática estendida

Importação

```
math <- fat.extra.mathex
```

Recomendação

Importe esta biblioteca em vez de `fat.math`, como `math <- fat.extra.mathex` pois ela inclui as [demais funções](#) matemáticas automaticamente.

Métodos

Nome	Assinatura	Breve descrição
fact	(x: Number): Number	Retorna o fatorial de x
logN	(x: Number, base: Number): Number	Retorna o logaritmo de x na base definida
exp	(x: Number): Number	Retorna e elevado à potência de x
tan	(x: Number): Number	Retorna a tangente de x
sinh	(x: Number): Number	Retorna o seno hiperbólico de x
cosh	(x: Number): Number	Retorna o cosseno hiperbólico de x
tanh	(x: Number): Number	Retorna a tangente hiperbólica de x
radToDeg	(r: Number): Number	Converte radianos para graus
degToRad	(d: Number): Number	Converte graus para radianos
mean	(v: List/Number): Number	Retorna a média de um vetor
median	(v: List/Number): Number	Retorna a mediana de um vetor
sigma	(v: List/Number): Number	Retorna o desvio padrão de um vetor
variance	(v: List/Number): Number	Retorna a variância de um vetor
sigmoid	(x: Number): Number	Retorna o sigmoid de x
relu	(x: Number): Number	Retorna o ReLU de x
round	(x: Number): Number	Retorna o inteiro mais próximo de x

Exemplo

```
math <- fat.extra.mathex # importação nomeada
math.fact(5)             # retorna 120
```

Veja também

- [Biblioteca Math](#)
- [Pacote Extra](#)

Memo

Memo

Classe de utilidade de memoização genérica

Importação

```
_ <- fat.extra.Memo
```

Construtor

Nome	Assinatura	Breve descrição
Memo	(method: Method)	Cria uma instância Memo para um método

Membros do Protótipo

Nome	Assinatura	Breve descrição
asMethod	() : Method	Retorna uma versão curried do Memo
call	(arg: Any): Any	Chamada memoizada; armazena e retorna resultado

Exemplo

Memo é útil para otimizar funções, armazenando os resultados. Ela armazena o resultado das chamadas de função e retorna o resultado armazenado quando as mesmas entradas ocorrem novamente.

```
_ <- fat.extra.Memo

fib = (n: Number) -> {
  n <= 2 => 1
  _      => quickFib(n - 1) + quickFib(n - 2)
}

memoInstance = Memo(fib)
quickFib = memoInstance.asMethod

quickFib(50) # 12586269025
```

Agora você pode chamar `quickFib` como se estivesse chamando `fib`, mas com resultados armazenados em cache para entradas computadas anteriormente.

aviso: pode causar acúmulo na alocação de memória

regex

Padrões comuns de expressão regular

Importação

```
_ <- fat.extra.regex
```

Constantes

- alphaOnly
- alphaNum
- digitsOnly
- emailAddress
- httpUrl
- ipAddress
- isbnCode
- numericValue

Notas de uso

Aqui estão alguns exemplos de como combinar o texto com as expressões regulares fornecidas por esta biblioteca usando o método `match` do protótipo `Text`:

```
_ <- fat.extra.regex

"abc".match(alphaOnly) # saída: true

"abc123".match(alphaNum) # saída: true

"123".match(digitsOnly) # saída: true

"johndoe@example.com".match(emailAddress) # saída: true

"https://www.example.com/page?query=param".match(httpUrl) # saída: true

"192.168.0.1".match(ipAddress) # saída: true

"1-56619-909-3".match(isbnCode) # saída: true

"3.14159e-5".match(numericValue) # saída: true
```

Observe que as expressões regulares podem exigir modificações com base em casos de uso ou requisitos específicos. Por exemplo, pode ser necessário modificar a expressão regular `httpUrl` para corresponder a URLs que incluem um número de porta. Certifique-se de testar seus próprios dados de entrada para garantir que estejam funcionando conforme o esperado.

No momento, o suporte `regex` do `FatScript` está limitado apenas à correspondência. Você não pode usar expressões regulares para operações de localizar e substituir.

Detalhes técnicos

As expressões regulares podem ser ferramentas muito poderosas, mas também podem ser complexas e difíceis de fazer direito.

O `FatScript` implementa o dialeto [POSIX regex estendido](#), que é o mesmo dialeto usado por `grep`. Nos bastidores, o `FatScript` usa a função [`regexec`](#) para executar a correspondência de expressões regulares.

Aqui está a implementação exata do `regex` fornecida por esta biblioteca que pode servir de inspiração para sua própria escrita:

```
# alphaOnly: corresponde apenas a um ou mais caracteres do alfabeto
^[:alpha:]+$

# alphaNum: corresponde a um ou mais caracteres do alfabeto e dígitos
^[:alnum:]+$
```

digitsOnly: corresponde apenas a um ou mais dígitos
`^[[:digit:]]+$`

emailAddress: corresponde a um endereço de e-mail válido, com um ou mais
 # caracteres alfanuméricos, pontos, sublinhados, sinais de adição ou hifens
 # antes do símbolo @, e um ou mais caracteres alfanuméricos, pontos ou hifens
 # após o símbolo @ seguido por um domínio superior de duas a quatro letras
`^[[:alnum:]]_.-]+@[[:alnum:]]_.-]+\.[[:alpha:]]{2,4}$`

httpUrl: corresponde a um URL válido http ou https, nome de domínio (um ou
 # mais caracteres alfanuméricos seguidos de um ponto) e o caminho (zero ou mais
 # caracteres incluindo caracteres alfanuméricos, pontos, hifens, pontos de
 # interrogação, sinais de igual, e comercial, porcentagem ou libra)
`^(http|https):\/\/([[:alnum:]]+\.)+[[:alpha:]]{2,6}([\/[:alnum:]]\.\.-\?=\&%#)+)?$`

ipAddress: corresponde a um endereço IP válido na notação dotted-quad,
 # com quatro grupos de um a três dígitos separados por pontos
`^([[:digit:]]{1,3}\.){3}[[:digit:]]{1,3}$`

isbnCode: corresponde a um ISBN com 10 ou 13 dígitos, com o último dígito
 # sendo um dígito de 0-9 ou a letra "X" para representar 10, permitindo hifens
 # ou espaços para serem usados como separadores entre grupos de dígitos
`^[0-9]{1,5}[-]?[0-9]{1,7}[-]?([0-9]{1,6}[-]?[0-9][0-9]([-]?[0-9]{3,5})[-]?[0-9X])$`

numericValue: corresponde a um valor numérico, incluindo um sinal negativo opcional
 # no início, um ou mais dígitos antes de uma parte decimal opcional (um ponto seguido
 # por um ou mais dígitos) e uma parte exponencial opcional (letra 'e' seguida por um
 # sinal opcional e um ou mais dígitos)
`^-?[[:digit:]]+(\.[[:digit:]]+)?(e[+-]?[[:digit:]]+)?$`

Ao definir expressões regulares em FatScript, prefira usar [textos brutos](#) e lembre-se de escapar as barras invertidas conforme necessário, garantindo que as expressões regulares sejam interpretadas corretamente.

Veja também

- [Pacote extra](#)

Sound

Sound

Interface de reprodução de som

Wrapper para players de áudio de linha de comando usando [fork e kill](#).

Importação

```
_ <- fat.extra.Sound
```

Construtor

O construtor `Sound` recebe três argumentos:

- **path**: o caminho do arquivo de áudio.
- **duração** (opcional): o tempo de espera (em milissegundos) para poder reproduzir novamente o arquivo. Geralmente, você deseja definir isso como a duração exata do seu áudio.
- **player** (opcional): o player padrão utilizado é `aplay` (utilitário de áudio comum no Linux, apenas dá suporte a wav), mas você pode usar `ffplay` para reproduzir mp3, por exemplo, definindo `ffplay = ['ffplay', '-nodisp', '-autoexit', '-loglevel', 'quiet']` e fornecendo-o como argumento para sua instância de som. Neste caso o pacote `ffmpeg` precisa estar instalado no sistema.

Membros do protótipo

Nome	Assinatura	Descrição
<code>play</code>	<code>()</code> : Void	Inicia reprodução, se ainda não estiver tocando
<code>stop</code>	<code>()</code> : Void	Interrompe reprodução, se ainda estiver tocando

o estado de "estar tocando" é inferido do parâmetro de duração

Exemplo

```
_ <- fat.extra.Sound
time <- fat.time

applause = Sound('applause.wav', 5000);
applause.play
time.wait(5000)
```

note que `Sound` cria um processo filho para reproduzir o áudio, portanto, a reprodução é assíncrona

util

util

Outras utilidades aleatórias

Importação

```
_ <- fat.extra.util
```

Constantes

- regex, importação nomeada da [biblioteca regex](#)

Métodos

Nome	Assinatura	Breve descrição
inferType	(x: Text): Any	Converte texto em booleano, número ou mantém como texto
fillWith	(fill: Text, n: Number): Text	Criar texto com n de elementos de preenchimento
padRight	(x: Text, fill: Text, n: Number): Text	Adiciona até n elementos de preenchimento ao lado direito de x
padLeft	(x: Text, fill: Text, n: Number): Text	Adiciona até n elementos de preenchimento ao lado esquerdo de x

Notas de uso

preenchimento de texto

Métodos para fins de formatação/preenchimento de texto, use assim:

```
padRight('Label', ' ', 10) # Text: # Text: 'Label      '
padLeft('45', '0', 6)     # Texto: '000045'
```

se o tamanho do texto for maior que n, nenhuma transformação é executada

Veja também

- [Biblioteca csv](#)
- [Pacote extra](#)

xml

xml

Analisador e gerador de XML simplificado

Importação

```
_ <- fat.extra.xml
```

[pacote de tipos](#) é automaticamente importado com esta importação

Variáveis

Essas configurações podem ser ajustadas para configurar o comportamento das funções de processamento:

- `showParseWarnings`, padrão é `true` - configure para `false` para suprimir avisos durante a análise

Métodos

Nome	Assinatura	Breve descrição
<code>toXml</code>	<code>(node: Any, wrap: Text = ∅): Text</code>	Gera xml a partir de tipos nativos
<code>fromXml</code>	<code>(text: Text): Any</code>	Converte xml para tipos nativos

Uso

toXml

Código de exemplo:

```
data = {
  bookstore: [
    { book: { title: 'Book 1', author: 'Author 1' } }
  ]
}
```

```
xmlString = toXml(data)
# xmlString será a representação xml dos dados
```

`toXml` gera string xml a partir de estruturas de dados `FatScript`

fromXml

Código de exemplo:

```
_ <- fat.extra.xml
```

```
xmlData = '<bookstore><book><title>Book 1</title><author>Author 1</author></book></bookstore>'
```

```
parsedData = fromXml(xmlData)
# parsedData será um Scope contendo os dados xml analisados
```

`fromXml` não suporta atributos ou tags auto-fechadas e mostrará avisos se `showParseWarnings` estiver configurado para `true`

listas são automaticamente inferidas quando múltiplos itens irmãos estão presentes, o que pode levar a estruturas de dados inconsistentes em casos onde um elemento é esperado ser uma lista, mas ocasionalmente contém apenas um único item

Veja também

- [Pacote extra](#)

Comandos embutidos

Comandos embutidos

Comandos embutidos são funções de baixo nível do FatScript que podem ser invocadas com palavras-chave precedidas por um cifrão \$. Esses comandos estão sempre disponíveis, implementados como código compilado e não requerem importações.

Ao contrário dos métodos, eles não recebem argumentos explícitos, mas podem ler a partir de nomes de entrada específicos no escopo atual ou até mesmo do estado interno do interpretador.

Os mais úteis

Aqui estão alguns comandos embutidos que podem valer a pena conhecer:

- `$debug` alterna os logs de depuração do interpretador
- `$exit` encerra o programa com o código fornecido
- `$keepDotFry` mantém a config (`.fryrc`) no escopo após a inicialização
- `$result` alterna a impressão do resultado no final da execução
- `$root` fornece uma referência ao escopo global
- `$self` fornece uma referência própria ao escopo do método/instância
- `$bytesUsage` retorna o total de bytes alocados no momento
- `$nodesUsage` retorna o total de nós alocados no momento
- `$isMain` verifica se o código está sendo executado como principal ou módulo

Você pode chamá-los diretamente no seu código, assim:

```
$exit # encerra o programa
```

para usar outros comandos embutidos você deve estudar a implementação C de `fry`, já que a lista completa não está documentada, consulte o arquivo [embedded.c](#)

Bibliotecas por trás dos bastidores

As bibliotecas padrão embalam chamadas embutidas como métodos, fornecendo uma interface mais ergonômica. Você não precisa criar um escopo de execução ou carregar argumentos nesse escopo antes de delegar a execução a eles.

Por exemplo, veja como você pode usar o método `floor` da [biblioteca math](#):

```
_ <- fat.math
floor(2.53)
```

Este método é implementado como:

```
floor = (x: Number): Number -> $floor
```

Por trás dos bastidores, o método `floor` cria um escopo de execução e carrega um argumento como `x` nele. O método então delega a execução ao comando embutido `$floor`, que por sua vez, lê o valor de `x` do escopo atual e retorna o menor inteiro que não é maior que este número.

Você pode obter o mesmo resultado que o método acima fazendo o seguinte:

```
x = 2.53
$floor # lê o valor de x do escopo atual
```

Hackeando

Você pode ver qual comando embutido um método de biblioteca está chamando, olhando para a implementação da biblioteca através do método `readLib` da [biblioteca sdk](#). Tecnicamente, não há nada que impeça de chamar comandos embutidos diretamente.

Por exemplo, você pode encerrar seu programa chamando `$exit` diretamente, que sairá com o código 0 (padrão) ou, se uma entrada numérica chamada `code` existir no escopo atual, o valor dessa entrada será usado como código de saída. No entanto, seria mais elegante importar a biblioteca `fat.system` e chamar o método `exit` com o código de saída desejado:

```
sys <- fat.system  
sys.exit(0) # exits with code 0
```

Essa abordagem torna seu código mais legível e menos propenso a erros, além de fornecer uma melhor separação de responsabilidades.

É importante ter em mente que os comandos embutidos são caixas pretas e não são destinados à escrita de código FatScript comum. Na maioria dos casos, você precisaria ler a [implementação em C subjacente](#) para entender melhor o que um comando está realmente fazendo.

Embora seja possível usar comandos embutidos para obter desempenho adicional em tempo de execução, evitando importações e chamadas de método, isso não é recomendado devido à perda de legibilidade do código. Em geral, é melhor usar as bibliotecas padrão e seguir as melhores práticas para escrever um código claro, fácil de manter.